

MNKW01

金 沢 大 学 計 算 機 セ ン タ ー

利 用 の 手 引 き

S A S 図 形

— S A S を 利 用 し た 簡 単 な 図 形 作 成 法 —

執 筆 者 工 学 部 助 手

中 野 康 英

はじめに

SAS^{*1}とは、数行のプログラムで、データの高度な統計処理やプロットをしてくれる、便利でか
しこいシステムである。

SASについては、既に利用の手引き特集 SAS 編が発行されており、多くのユーザが活用してい
る。SASによる図形作画も、既に多くのユーザが利用しているが、既刊の利用の手引きは、統計処
理機能を中心にして解説されており、SASによる図形作画についてまとめられたものが無かった。
そのため新たにSASを利用して図形を作画してみようとする初心ユーザや、更に別の作画にチャレ
ンジしてみようとするユーザには、若干の障害があった。

そこで、この手引きではSASの図形作画諸機能のうちから、比較的良好に利用されと思われる機
能を選び出してまとめてみる。初心ユーザやチャレンジユーザの参考になれば幸いである。

また、これまでの図形出力は、殆どNLP^{*2}に限られていたので、工学部ユーザ以外には不便さがあ
った。しかし最近、OPR^{*3}およびグラフ機能付の日本語端末へも出力できるようになったので、
それらの実行法を紹介する。また、これまで使用できなかったカラー機能も使用可能となったので、
合わせて紹介する。なお、図形作画と統計処理との共通部分については説明を割愛するので、利用の
手引き特集 SAS 編を参照されたい。

この手引きでは、まず最初にSASに共通な、SASプログラムの構成や基本文法および実行方法
等について述べる。その後第3章で、実際の出力図形と共に例題を掲載する。各ユーザは第1章を読
んだら第3章に飛び、自分のやりたいことに近い例題だけをつまみ喰いすると良いと思う。その便利
のために、第3章の最初でこの手引きで紹介している主な機能についてまとめて掲載する。なお、そ
れぞれの図形とそれを作画したSASプログラムは図番号とプログラム番号で対応している。SAS
プログラムの原形ができたところで、第2章へ戻って適当な実行方法を見つけて実行する。必要なら
ば更に第3章へ戻ってグラフを工夫すれば良い。

なお、この手引きに掲載したすべてのプログラム、および作画に必要なデータは、ユーザが自由に
複写して使用できる。その方法については付録7を参照されたい。

FOOTNOTE

*1 (Super Application System) 機能の拡大に伴って以前の
Statistical Analysis System から読み換えた。

*2 (Nihongo Line Printer) 日本語ラインプリンタ。センター内にある。

*3 (Office PRinter) オフィスプリンタ。センターおよび、城内分室、宝町分室
にある。

1 . SAS プログラムの構成と基本文法 *1

1 . 1 SAS プログラムの構成

最も単純な図形作画用プログラム例とその出力結果をプログラム 1 と図 1 に示す。このプログラムは、ファイル DDD に結合されたデータセット *2 から X, P データの組を読み込んで、プロットするものである。この例からわかるように、SAS プログラムは非常に簡単であり、わずか数行のプログラムで、ともかくも図は作ってくれる。但し、SAS プログラムの言語は FORTRAN ではない。従って、GRACE *3 や GSP *3 のように FORTRAN プログラム内に組み込むことはできない。

```
DATA; INFILE DDD; INPUT X P;  
GOPTIONS DEVICE=SASPSP;  
PROC GPLOT; PLOT P*X;
```

プログラム 1

また、SAS プログラムは FORTRAN などによるプログラムに比べれば、はるかに単純な形をしている。プログラムの内容は、次の 2 つに大別される。*4

- 1) DATA ステップ … 処理すべきデータを準備する。(プログラム 1 の 1 行目)
- 2) PROC ステップ *5 … データを処理する。(プログラム 1 の 3 行目)

FOOTNOTE

- *1 1.1, 1.2 の殆どを文献 2 の 1.1 から引用した。
- *2 ファイル名 DDD はユーザが適当に与える名前であり、プログラム 1 行目の INFILE ステートメントで指定している。このファイルへのデータセットの結合は、プログラムの実行時に行う。
- *3 図形作画用サブルーチン集。それぞれ文献 8)、9) 参照。
- *4 プログラム 1 の 2 行目の GOPTIONS ステートメントは、どちらのステップ内にも置くことができる。従って 2 行目はどちらのステップと考えてもよい。
- *5 PROCedure の略であるが、いつも頭文字 4 文字のみで使用される。




図1 最も単純なプログラムによる作画例（縮尺1.0）

これらのステップは、同一の SAS プログラム内で何回使用してもよい（プログラム 9-2、プログラム 24-3 参照）。

1.2 SAS プログラムの基本文法

SASプログラム内の全てのステートメント*6は、必ずセミコロン(;)で終わる。

従って、ステートメントは1行に複数個書いてもよいし、また1つのステートメントが何行にわたってもよい。但し、単語の途中で改行されてはいけない。

変数の取り扱い等はFORTRANとほぼ同じである。

但し、配列を意識的に用意する必要はない。1つの変数名でデータを何回も読み込めば、勝手に配列としてくれる。

FOOTNOTE

*6 (Statement) プログラムから計算機に伝達する命令の一区切り。

1.3 DATA ステップについて

この節は、図形作画処理、統計処理に共通な部分であり、文献2の第2章に、多くの例を挙げて、ていねいに説明されているので、そちらを参照していただきたい。

ここでは大まかに、どんなことができるかについて述べるにとどめる。

DATA ステップは、処理すべきデータを準備するステップであるが、データの受け取り方には、次の3方法がある。

- 1) SAS プログラムに直接書き込む(プログラム2-2および文献2の2.4のCARDSステートメント参照)。
またはSAS プログラム内で計算する(プログラム16および文献2の2.7のプログラミング・ステートメントとSAS 関数参照)。
- 2) 他のデータセットから読み込む(プログラム2および文献2の2.5のINFILEステートメント参照)。
- 3) 既に登録済のSAS データセットを使う(プログラム18および文献2の第5章参照)。

この手引きでは、殆んど2)の方法をとっているが、各ユーザは使用目的にかなったものを選択して用いれば良い。それによって次のPROC ステップが影響を受けることはない。

また、DATA ステップでは、受け取ったデータを加工することができる。そのために、代入ステートメントやDOステートメント、IFステートメントなどを使用することができる。またいくつかの関数も準備されている。詳しくは、文献2の2.7プログラミング・ステートメントとSAS 関数を参照されたい。^{*7}

幾つかのものは、この手引きの例題でも用いている。(例えばプログラム16)。

1.4 PROC ステップについて

PROC ステップは、実際にデータを処理するステップである。処理の種類別に名前がついておりプロシジャ名という。この手引きで紹介するプロシジャは、表1-1の1.、2.、3.、の3つの図形作画プロシジャと、4.のGCONTOUR およびG3Dの前処理を行うプロシジャ、および5.の文字や記号作画用のプロシジャの合計5個である。^{*8} 他にも図形作画用のプロシジャはあるが、ここでの紹介は割愛する。文献4を参照されたい。

FOOTNOTE

*7 このステップは、文献2でも説明されていない機能もまだいくつか持っている。本格的にSAS を利用されるユーザには、文献5の2~6章や文献6のⅧを読まれることをお勧めする。

表1-1 この手引きで紹介されるプロシジャ

プロシジャ名	機能	例
1. GPLOT	2変数間のグラフの作成。	2~15
2. GCONTOUR	等高線図の作成。	16~18
3. G3D	3次元鳥瞰図の作成。	19~23
4. G3GRID	等高線図または3次元鳥瞰図用のデータが直交格子状でない時に、補間によって直交格子状に変換する。	24
5. GSLIDE	文字・記号のみの作画。	25

表1-2 この手引きで紹介されるグローバル・ステートメント

ステートメント名	機能	例
1. GOPTIONS	図形作画に関するオプションの指定。	全部
2. TITLE, FOOTNOTE, NOTE	図形の上または下にタイトルを書く。	4~6 25等
3. LABEL	変数名に文字列を対応させる。	4~6等
4. SYMBOL	GPLOTプロシジャでの種々のオプションの指定。	6~16
5. PATTERN	特定の閉領域にハッチングを行うための指定。	15, 18

また、プロシジャではないが、表1-2の5種のステートメントは図形作画でよく利用されるので、合わせて紹介する。^{*8} なお、これらのステートメントは、どちらのステップに置いても良い。^{*9}

FOOTNOTE

*8 付録では SORT プロシジャおよび BY ステートメントについても紹介する。これらは、特定の変数の値毎に同一の処理を繰り返し行う場合に、用いると便利である。

*9 このようなステートメントを、グローバルステートメント (Global statement) という。これら以外のステートメントは、どちらかのステップでしか使用できない。

2 . S A S プ ロ グ ラ ム の 実 行 と 出 力 方 法

SAS プログラムがデータセット AAA.DATA (BBB) に入っており、作画用データをデータセット XXX.DATA (YYY) から読み込んで、作画することを考えよう。

なお、ここで AAA や XXX のように、同じ文字を3つ重ねて書いてある部分は、ユーザが自由に指定できる名前である。以後も同様の表記法を用いる。

これを実行するには、いくつかの方法があるが、比較的一般的と考えられるのは、次の3つの方法である。

- (1) TSS で実行し、作画結果を OPR に出力する。
- (2) TSS で実行し、作画結果を NLP に出力する。
- (3) グラフ機能付日本語端末で実行し、端末画面上に作画させる。

以後、この3つの方法と、カラー機能の使用法について説明する。

また、上記の3方法によるプログラムとコマンドの違いを表2-1に示す。

2 . 1 T S S で 実 行 し、O P R に 出 力 す る

SAS 実行のためのコマンドは次のようである。

```
SP □ AAA.DATA (BBB) □ F (DDD) □ DA (XXX.DATA (YYY)) □ OPR (OC01)
```

但し、SAS プログラム中の GOPTIONS ステートメント中の DEVICE= には SASPS を指定しなければならない。

SP は、SAS プログラムを実行するためのコマンドである。いつもこのまま書く。

AAA.DATA (BBB) は、SAS プログラムが保存されているデータセット名である。各ユーザが適当なものを指定しなければならない。

F (DDD) □ DA (XXX.DATA (YYY))

は、作画用データを他のデータセットから読み込む場合に必要なものである。

F (DDD) の F () はキーワードであり、いつもこのまま指定する。() 内の DDD には、SAS プログラムの INFILE ステートメントで指定したファイル名を入れる。例えば、プログラム1の場合には、上のように F (DDD) となる。

DA (XXX.DATA (YYY)) の DA () はキーワードであり、いつもこのまま指定する。() 内には作画用データの保存されているデータセット名を指定する。^{*1}

OPR (OC01) は、OPR に出力する場合に必要なオペランドである。OPR () はキーワードであり、いつもこのまま指定する。() 内の OC01 は OPR 名^{*2} であり、最寄のものを指定すれば良い。

上記コマンドを実行すると、エラーメッセージなどのいくつかのメッセージが表示され、作画結果は、直ちに指定の OPR に出力される。特別な出力要求はいらない。

図 1-2 に正常なメッセージの例を示す。またエラーメッセージの例を図 1-3, 図 1-4 に示す。図 1-3 は、データを読み込んでいない変数 Y を誤って使用したものであり、図 1-4 は、2 行目の最後の ; を落としてしまったものである。このように SAS は、エラーの見つかった場所にアンダーラインをつけ、その下にエラー番号をつけてくれる。エラー番号についての説明はその下につけられている。

```
ENTER(OPR) DATA SET(PS) NAME
ZZZ
NOTE: SAS RELEASE 82.3 AT KANAZAWA UNIVERSITY (02458001).
NOTE: CPUID   VERSION = 30   SERIAL = 000826   MODEL = 0360 .
      CPUID   VERSION = 30   SERIAL = 000827   MODEL = 0360 .

NOTE: INFILE DDD(DATA1) IS:
      DSNAME=AB9999.XXX.DATA(YYY),
      UNIT=DISK,VOL=SER=USER01,DISP=OLD,
      DCB=(BLKSIZE=3120,LRECL=80,RECFM=FB)
NOTE: 15 LINES WERE READ FROM INFILE DDD(DATA1).
NOTE: DATA SET WORK.DATA1 HAS 15 OBSERVATIONS AND 2 VARIABLES. 1000 OBS/TRK.

NOTE: KKA/TOKYO-034370921/OSAKA-064457561/NAGOYA-0522115021
```

図 1-2 正常なメッセージの例

FOOTNOTE

- * 1 なお、作画用データのデータセットも SAS プログラムのデータセットも共にデータセットタイプは DATA で良いから、データセットの数を増さないためには、同じデータセットを用いてメンバ名で区別すれば良い。その場合は、上記コマンドの AAA.DATA と XXX.DATA には、当然同じものを指定することになる。
- * 2 各 OPR に名前が貼り付けてある。


```

ENTER(OPR) DATA SET(PS) NAME
ZZZ
NOTE: SAS RELEASE 82.3 AT KANAZAWA UNIVERSITY (02458001).
NOTE: CPUID   VERSION = 30   SERIAL = 000826   MODEL = 0360 .
      CPUID   VERSION = 30   SERIAL = 000827   MODEL = 0360 .

NOTE: INFILE DDD(DATA1) IS:
      DSNAME=AB9999.XXX.DATA(YYY),
      UNIT=DISK,VOL=SER=USER01,DISP=OLD,
      DCB=(BLKSIZE=3120,LRECL=80,RECFM=FB)
NOTE: 15 LINES WERE_READ FROM INFILE DDD(DATA1).
NOTE: DATA SET WORK.DATA1 HAS 15 OBSERVATIONS AND 2 VARIABLES. 1000 OBS/TRK.

      3 PROC Gplot; PLOT Y*X;
      -
      155
155: THE VARIABLE NAME IS NOT ON THE DATA SET.
NOTE: SAS STOPPED PROCESSING THIS STEP BECAUSE OF ERRORS.

NOTE: KKA/TOKYO-034370921/OSAKA-064457561/NAGOYA-0522115021

```

図1-3 エラーメッセージを含む例

OPR への出力に当っては、特に指定がなければ、A4 サイズの用紙が横長に用いられる。B4 サイズの用紙に出力したい時および用紙を縦長に使用したい時には、それぞれ次のように、必要なオペランドを追加する。

B4 サイズ用紙への出力

```
SP ..... OPR (OC01) _B4
```

縦長用紙への出力

```
SP ..... OPR (OC01) _LA
```

このオペランドの両方を追加すれば、B4 用紙を縦長に使用して出力される。

なお、B4 の代わりに A4 と指定することもできる。^{*3} 但し、どちらの指定もなければ A4 が指定されたものとみなされる。また、LA の代わりに PO (横長用紙) と指定することもできる。^{*4} 但し、どちらの指定もなければ PO が指定されたものとみなされる。

FOOTNOTE

*3 それぞれの出力範囲は B4 が 34.5 × 24.1cm、A4 が 27.9 × 19.6cm である。

*4 LA は Landscape (風景画) の略、PO は Portrait (肖像画) の略であると思われるが、反対に対応しているので注意を要する。

```

ENTER(OPR) DATA SET(PS) NAME
ZZZ
NOTE: SAS RELEASE 82.3 AT KANAZAWA UNIVERSITY (02458001).
NOTE: CPUID   VERSION = 30   SERIAL = 000826   MODEL = 0360 .
      CPUID   VERSION = 30   SERIAL = 000827   MODEL = 0360 .

      3 PROC GPLOT; PLOT P*X;
      -----
      603 603   180
180: STATEMENT IS NOT VALID OR IT IS USED OUT OF PROPER ORDER.
603: INVALID STATEMENT PARAMETER.
NOTE: INFILE DDD(DATA1) IS:
      DSNAME=AB9999.XXX.DATA(YYY),
      UNIT=DISK,VOL=SER=USER01,DISP=OLD,
      DCB=(BLKSIZE=3120,LRECL=80,RECFM=FB)
NOTE: SAS STOPPED PROCESSING THIS STEP BECAUSE OF ERRORS.
NOTE: SAS SET OPTION OBS=0 AND WILL CONTINUE TO CHECK STATEMENTS.
      THIS MAY CAUSE NOTE: NO OBSERVATIONS IN DATA SET.
NOTE: 0 LINES WERE READ FROM INFILE DDD(DATA1).
NOTE: DATA SET WORK.DATA1 HAS 0 OBSERVATIONS AND 2 VARIABLES. 1000 OBS/TRK.

NOTE: KKA/TOKYO-034370921/OSAKA-064457561/NAGOYA-0522115021

```

図1-4 エラーメッセージを含む例

2.2 TSSで実行し、NLPに出力する

SAS 実行のためのコマンドは2.1の場合とほぼ同じである。

SP _ AAA.DATA(BBB) _F(DDD) _DA(XXX.DATA(YYY)) _PLOT(NLP)

但し、SAS プログラム中の GOPTIONS ステートメント中の DEVICE=には SASNLP と指定する。なお、これまで PSPLOT と指定していたユーザも、SASNLP と変更すること。^{*5}

PLOT(NLP) は、NLP に出力する場合に必要なオペランドである。() 内も含めていつもこのまま指定する。

FOOTNOTE

*5 PSPLOT 指定は、今後、確実に機能するとは限らない。

このコマンドを実行すると

```
ENTER (NLP) _ DATA _ SET (PS) _ NAME
```

と表示され、NLP 出力用データセット名の入力要求があるので、適当な名前を入力する。但し、データセット・タイプは勝手に OUTLIST となるので、データセット・タイプを入力する必要はない。

例えば

```
ZZZ
```

と入力すると、2.1と同様にいくつかのメッセージが表示され、正常に実行されれば作画結果は、ZZZ.OUTLIST というデータセットに書かれ、出力要求を待っている。

これを実際に NLP に出力するためのコマンドは、次のようである。

```
PSPNLP _ ZZZ . OUTLIST
```

そうすれば、いくつかのメッセージが表示された後、作画図形は NLP に出力される。トークンカードによる出力要求はいらない。

NLP へ出力する場合の実行法および出力要求は、この他にもさまざまある。それらについては付録6にまとめて示す。

2.3 グラフ機能付日本語端末で実行し、端末画面上に作画する

このためのコマンドは、次のようである。

```
SP _ AAA . DATA (BBB) _ F (DDD) _ DA (XXX . DATA (YYY))
```

追加オペランドはいらない。但し、SAS プログラム中の GOPTIONS ステートメント中の DEVICE= には F6653 と指定する。

上のコマンドを実行すると、いくつかのメッセージ表示後、画面は一旦消去され作画が始まる。作画終了の明瞭な合図はない。画面左下の〔X SXSTEM〕表示が消えれば作画終了である。

〔PRINT〕キーを押せば、近くの BGP^{*6} に画面のハードコピーが得られる。また、〔ENTER〕キーを押せば、画面は消去され、READY 状態に戻る。^{*7}

この日本語端末上への作画は、他の方法に比べて作画結果が早く見られる^{*8} という特徴がある。

表2-1 作画結果の出力先によるプログラムおよびコマンドの違い

出力先		OPR	NLP	日本語端末
プログラムの GOPTIONS ステートメント	DEVICE VSIZE HSIZE	SASPSP cm単位 "	SASNLP cm単位 "	F6653 インチ単位 "
作画最大範囲		(A4)27.9×19.6cm (B4)34.5×24.1cm	34×25cm	10.5×8.5in
実行コマンドの追加オペラント		OPR (OPR名)	PLOT (NLP)	必要なし
図形出力要求コマンド		必要なし	PSPNLP	ハードコピーのみ

2.4 カラー機能の使用法

端末は、BGD^{*9}またはCAD 端末^{*10}のカラー機能付グラフィックディスプレイを用いなければならない。

実行のためのコマンドは2.3と同じであり、PLOT オペラントはいらない。SAS プログラム内のDEVICE指定も2.3と同じであり、F6653と指定する。

使用可能な色は、青、赤、緑、黄、ピンク、シアン、白の7色である。プログラムの例を付録5のプログラムE、F、G、Hに示す。詳細は文献4を各自勉強されたい。

なお、CAD 端末の場合は画面のハードコピーをカラープリンタに出力することができる。

FOOTNOTE

- * 6 (Business Graphic Printer) OPR のことではない。日本語出力の可能なプリンタである。
- * 7 複数枚の図を書かせている時は、次の図の作画に移る。
- * 8 グラフィックディスプレイ上への作画よりも作画速度が速い。
- * 9 (Business Graphic Display) センターに2台ある。
- * 10 センター内の特殊入出力室にある。

2.5 その他の実行法

上で紹介した他の方法については、次の条件に応じて適当なものを選択することとなる。

- ① データの受け取り方 … 2.3で述べた3方法により異なる。詳細は文献2の4.2および5.3を参照されたい。この手引きの3.2と3.17、3.18に例を示す。
- ② どこへ出力するか … 上に紹介したものの他に、図形検索用データセット、オフラインプロットに出力できる。詳細は文献1の第3章を参照されたい。なお、出力先によりSASプログラム内のDEVICE=の名称も変更しなければならないので注意されたい。
- ③ 既存プログラムの実行か、会話的な実行か……これまでのように図形出力までのプログラムが予め作製されていなくても、プログラムを少しずつ実行させながら処理していくこともできる。但し処理プログラムは一時的なものであり、データセットには残らない。この方法については、文献2の4.2を参照されたい。プログラム1程度の簡単なものならば、一時的プログラムでちょっと作製すれば、簡単にプロットが得られて便利であろう。
- ④ TSSによる実行かバッチによるか
これまではTSSによる実行について述べて来た。バッチによる実行方法については、文献1の4.6を参照されたい。

3 . 例題による各プロシジャの解説

3 . 1 この章で作画する図形

「はじめに」でも述べたように、この章は全体を通読する必要はない。各ユーザは、自分のやりたい機能だけをつまみ喰いすれば良い。その便利のために、ここでは各図で取り上げられている主な機能について、まとめて示す。

図2～図5は、グラフの軸やタイトルについて

図6～図11は、プロットとその連結線および回帰線について

図12～図14は、繰り返しのあるデータの処理について

図15は、特定領域のハッチングについて

図16～図18は、等高線図について

図19～図23は、鳥瞰図について

図24は、非直方格子データによる等高線図および鳥瞰図について

図25は、文字および記号のみの作画について

である。

3 . 2 図のサイズ以外、すべて SAS まかせ

SAS プログラムをプログラム2に、その出力結果を図2に示す。極めて簡単なプログラムであるが、とにかくグラフは作ってくれる。これが SAS の1つの特徴である。なおプログラム2はプログラム1とほぼ同じである。

```
DATA; INFILE DDD; INPUT X P;  
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) NOCHARACTERS VSIZE=15 HSIZE=20;  
PROC GPLOT; PLOT P*X;
```

プログラム2




図2 図のサイズ以外すべてSASまかせ(縮尺0.7)

【プログラム2の説明】

1行目は DATA ステップである。

`DATA;` は、データステップの始まりを示すキーワードであり、通常は、このまま書けば良い。

`INFILE_DDD;` は、DDD という名前のファイルからデータを読み込めという指令。従って、ファイル DDD には、SAS プログラム実行時に、データの記憶されているデータセットを割り当てておかなければならない。これは、2.1 で示した SP コマンドの F (ファイル名) と DA (データセット名) で簡単に準備できる。

`INPUT_X_P;` は、ファイル DDD に割り当てられたデータセットから、データ X と P を読めという指令。但し、このようにフォーマットなしで変数の並びのみで読み込ませるには、データセットの1行には1組のデータしかないこと、および、変数の値の間は1個以上の空白が存在することなど、幾つかの規則がある^{*1}。通常は特別意識しないでもパスできる規則である。

このプログラムでは1行目のみで DATA ステップは終了しているが、SAS は、データ

セットにデータが存在する限り勝手に繰り返し DATA ステップを実行してくれる。その際、変数 X、P は、FORTRAN で言う配列のように、順次、データ値を記憶してくれる。特別な配列宣言はいらない。

2 行目は、図形を作画するために必要な指定である。

GOPTIONS はキーワードでいつもこのまま指定する。以下の **DEVICE=** や **VSIZE=** などは **GOPTION** ステートメントのオプションであり、オプションを含めて、1つのステートメントとなる。従ってオプションの最後に ; がつく。

DEVICE=SASPSP は、図形をどこに作画するかを指定するもので **DEVICE=** まではキーワードでいつもこのまま。

SASPSP は、**OPR** に出力する時に指定する。NLP 出力または図形検索の場合は、**SASNLP** と指定する。グラフ機能付日本語端末またはグラフィックディスプレイ上に作画する場合は、**F6653** と指定する。またオフラインプロッタに出力する場合は、**OFFPLOT** と指定する。

COLORS=(BLACK) は、通常のプロットには不要だが、鳥瞰図等には必要なので、カラー機能を使わない場合はこのまま指定しておくのが良い。通常のプロットでは邪魔にはならない。

NOCHARACTERS は、このまま指定しておいた方が良い。この指定をしないと、図形の縦横比が文字基準になってしまい、意図通りのサイズにならないことがある。

VSIZE=15_HSIZE=20 は、作画の縦サイズ、横サイズの指定である。単位は cm^{*2} である。これらの値の上限については、2.1 の脚注 *3 でふれた。

3 行目は、図形作画のための **GPlot** プロシジャである。

PROC_GPlot ; は、それぞれキーワードであり、通常このまま指定する。

PLOT_P*X ; は、P を縦軸に X を横軸にして作画せよという指定である。PLOT はキーワードである。その後ろに縦軸変数名 * 横軸変数名と指定する。

また、データを読み込むべきデータセットのメンバ名までいつも決っている場合には、例えば **XXX.DATA(YYY)** の場合には、**INFILE** ステートメントを次のように変更しても良い。

INFILE _DDD.YYY ;

それに伴って、実行コマンドも変わり、**DA()** の () 内にはメンバ名まで指定する必要はなくな

FOOTNOTE

* 1 文献 2 の 2.3 参照。

* 2 **DEVICE=F6653** とした場合は単位はインチである。その場合の作画サイズ上限は、**VSIZE=8.5, HSIZE=10.5** である。

る。即ち、次のようになる。

```
SP _ AAA . DATA (BBB) _ F (DDD) _ DA (XXX . DATA) _ OPR (OC01)
```

また、プロットすべきデータ点が少ない時は、わざわざ他のデータセットから読み込まなくても、直接プログラム内に書き込めば良い。その例をプログラム 2-2 に示す。INFILIST ステートメントが無くなり、そのかわり INPUT ステートメントの後に CARDS ステートメントが置かれ、その後にデータ値が X、P の組で書き込まれている。また GOPTIONS ステートメントの前に ; がつけられている。

なお、このようにデータを外からもらわない場合の実行には、F () オペランドおよび DA () オペランドはいらない。従って、次のようなコマンドとなる。

```
SP _ AAA . DATA (BBB) _ OPR (OC01)
```

```
DATA; INPUT X P; CARDS;
    0.1      28.4
    0.2      20.2
    0.3      14.3
    0.4      12.3
    0.5      9.62
    0.6      7.98
    0.7      7.15
    0.8      5.95
    1.1      5.28
    1.2      4.64
    1.3      4.48
    1.4      5.33
    2.0      3.27
    2.1      2.96
    2.2      2.78
;GOPTIONS DEVICE=SASPSP COLORS=(BLACK) NOCHARACTERS VSIZE=15 HSIZE=20;
PROC GPLOT; PLOT P*X;
```

プログラム 2-2

3.3 目盛値の文字サイズとプロットシンボルの基本サイズの調整

図2の SAS まかせの作画では、目盛値とプロットシンボルが小さ過ぎるので、これらの大きさの調整をする。

SAS プログラムをプログラム 3 に、その出力結果を図 3 に示す。プログラムは 3 行目が 1 行挿入されたに過ぎないが、シンボルサイズと文字サイズが変わっていることに注意されたい。

```

DATA; INFILE DDD; INPUT X P;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GPLOT; PLOT P*X;

```

プログラム 3




図3 目盛値の文字サイズとプロットシンボルの基本サイズの調整 (縮尺0.7)

【プログラム 3 の説明】

プログラム 2 に 3 行目が挿入されただけである。

`VPOS=30 HPOS=60` は、GOPTIONS ステートメントのオプションであり、文字の基本サイズとピッチを決める。^{*3} 即ちVSIZE, HSIZEで指定された縦横の長さをそれぞれ

FOOTNOTE

* 3 この部分の説明は、SAS 発行のマニュアル (例えば文献 4) と、少し異っている。この説明の方が適切であると思われる。

VPOS , HPOS で指定された数に分割した仮想の格子を作る。そして SAS の持っている幾つもの文字サイズ*⁴ のうちで、その格子に納まる最大の文字を用いて文字を作画する。また、この仮想格子が文字のピッチとなる。ところで、SAS 文字の縦横比は、およそ 3 : 2 と思われるので、仮想格子の縦横比がこの値より著しくずれて横長になると、ピッチの間のびした文字列となり、縦長になるとタイトルの行間が大きくなってしまふことになる。

なお、プロットシンボルは、特に指定しなければ基本文字サイズの 1/2 になる。

但し、上の説明は、NOCHARACTERS オプションが指定された場合であり、このオプション指定がされない場合には、文字サイズの決定は同様であるが、今度は逆に文字サイズを基本にして仮想格子を作り直してしまうので、文字の納まりは良くなるが、図全体のサイズは指定と異なってしまうことになるので、注意を要する。

3 . 4 目盛値の指定、参照線の位置と線種指定、軸のラベル指定、タイトルの字体指定および小文字の使用

図3のままでは、あまりにさみしいので、標題のような指定をして少しグラフを飾ってみる。SAS プログラムをプログラム4に、その出力結果を図4に示す。

【プログラム4の説明】

DATA ステップ、GOPTIONS ステートメントは、プログラム3と同様である。但し、VPOS, HPOS がそれぞれ 1.5 倍になっているので、基本文字サイズが小さくなっている。

4行目から8行目までがGPLOTプロシジャである。これまで1行で納まっていた PLOT ステートメントが長くなって4行にわたっている。

9~11行目はLABELステートメントと FOOTNOTE ステートメントであり、DATA ステップに置いても良い。

■ PLOTステートメントの説明 ■

PLOT ステートメントの一般形は、次のようである。

PLOT 縦横変数名*横軸変数名/オプションの並び;

プログラム4の例では5行目のVZEROから8行目の;までが、PLOT ステートメントのオプションである。オプション指定がない場合には、プログラム3などのように/も含めて省略できる。


FOOTNOTE

* 4 筆者は勉強不足で、具体的なサイズを示すことはできない。判明すれば書き加えたいので、どなたか探究してお知らせ願いたい。

ここで用いられているオプションは次のものである。

```
DATA; INFILE DDD; INPUT X P;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=45 HPOS=90;
PROC GPLOT;
  PLOT P*X/VZERO VAXIS=0 TO 30 BY 5
       HZERO HAXIS=0 TO 2.5 BY 0.5
       VREF=10 20 30          LVREF=3
       HREF=0.5 TO 2.5 BY 0.5 LHREF=8;
  LABEL X=DISTANCE (M) P=PRESSURE (ATM);
  FOOTNOTE1 .H=2.0 .F=TRIPLEX PRESSURE DECAY WITH DISTANCE;
  FOOTNOTE2 .H=1.5 .F=COMPLEX (Fマオン=Pネネアトオ);
```

プログラム 4



PRESSURE DECAY WITH DISTANCE (Fuel=Propane)

図4 目盛値の指定, 参照線の位置と線種指定, 軸のラベル指定,
タイトルの字体指定および小文字の使用 (縮尺 0.7)

VZERO

縦軸の目盛を0からスタートさせたい時に指定する。但し、次のVAXISオプションの指定がある場合は無視される。どちらのオプションも指定されていない場合は、縦軸の目盛はSASが適当に調整してくれるが、0からスタートするとは限らない。

VAXIS=目盛値の並び

縦軸の目盛を指定する。目盛値の並びは必ず昇順または降順に指定しなければならない。また、指定された値の間隔に関係なく目盛は等間隔に取られるので注意を要する。従って例のようにDO形式^{*5}を使って指定するのが安全である。

HZERO

HAXIS=目盛値の並び

これらはVZERO, VAXISと同様である。但し、横軸に関する指定である。

VREF=値の並び

縦軸に関する参照線を引きたい場合に指定する。値の並びはもちろん参照線を引きたい値の並びである。例のように実際に値を並べてもよいし、DO形式を用いても良い。また、これらを混在させてもよい。^{*6}

LVREF=線種のコード(短縮形LV=)

VREF オプションと共に指定して、VREFで指定した参照線に使う線種を指定する。線種は、図A(付録1)に示すコード番号で指定する。

HREF=値の並び

LHREF=線種のコード(短縮形LH=)

これらはVREF, LVREFと同様である。但し横軸に関する指定である。

FOOTNOTE

* 5 次の形式で指定する。

キーワード=初期値 TO 最大値 BY 間隔

数値は整数でも実数でも良い。初期値から最大値を超えないまでの等差数列を指定したことになる。

* 6 実際の数値の並びとDO形式による指定が混在した例は、プログラム17のLEVELS=の指定で使われている。

LABELステートメントの説明

軸にラベルをつけるときに指定する。省略すると、図3のように横軸の中心に横軸の変数名が、縦軸の上端に縦軸の変数名が書かれる。

LABELステートメントで、次のように指定すると

LABEL 縦軸変数名=縦軸ラベル 横軸変数名=横軸ラベル ;

それぞれの軸の中心にラベルが書かれる。但し、縦軸のラベルは縦書きになるのに注意されたい。ラベルの長さは40文字までである。^{*7} また、どちらか一方だけの変数について指定しても良い。

FOOTNOTEの説明

図の下側にタイトルを書きたい場合に指定する。

図の上側に書く場合はTITLEステートメントである。(3.25参照)

FOOTNOTE ステートメントの一般形は次のようである。

FOOTNOTE *n* オプション テキスト ;

FOOTNOTE はキーワードである。語尾の*n*は1から10までの数字であり、1行しかない場合は省略してもよい。*n*の最大のものが最下行に書かれる。また特に指定しなければ、図の横幅の中心に書かれる。

オプションの詳細は3.25で説明する。

ここで用いられているオプションは.H^{*8}と.F^{*8}である。

.H=文字のサイズ

基本文字サイズ以外の文字サイズを用いる場合に指定する。単位は基本文字サイズである。テキストの途中で指定しても良い。(3.25に例を示す)

.F=文字の種類

テキストの作図に用いる字種を指定する。図C(付録3)に使用可能な字種の例を示す。詳細については、文献4を参照されたい。但し、通常使用しそうな字種については、全てこの手引きの中で使用したつもりである。

テキストの指定法

オプションの指定の直後に空白を1つだけあけて指定する。それ以外の空白は全て有効な文字列の一部と見なされる。小文字を書くには、小文字用端末^{*9}を用いて実際に小文字を入力する

FOOTNOTE

*7 途中で改行する場合は注意を要する。文献4のP12～13を参照されたい。

*8 前に.(ピリオド)が付いているので注意されたい。

か、または通常の端末でカタカナに変換して入力する。小文字とカタカナの対応は次のようである。

a b c d e f g h i j k l m n o p q r s t u v w x y z
アイウエオカキクケタチツテトナニヌネヘホマミムメモヤ

3.5 縦軸ラベルの作画方向の変更、対数軸の採用、 目盛値間分割目盛線の指定

SAS プログラムをプログラム5に、その出力結果を図5に示す。

【プログラム5の説明】

4行目まではプログラム4と全く同じ。

5行目のVAXISオプションの目盛値の指定法が縦軸に対数軸を用いることの指定である。SASでは対数軸であることを明瞭に指定する方法はないので1オーダーだけの対数軸を設定することはできないように思われる。

縦軸ラベルの作画方向の変更は、次のように行っている。

まず、縦軸のラベル指定をやめる。その代わりにTITLEステートメントに次のオプションを指定して用いる。 .A=90

.A オプションの意味については3.25で説明する。

そしてタイトル文字列の前または後に空白を適当に加えて位置を調整する。^{*10}

目盛値間分割目盛線の指定は、HMINORないしVMINORオプションを用いて行う。

HMINOR=目盛値間目盛数

6行目のHMINORオプションは、HAXISオプションで指定した目盛値の間をさらに分割してつける目盛の数（分割数ではない）を指定する時に用いる。省略するとSASが勝手に行うので、図4のように、0.5間隔の目盛値間を10分割したりする。

VMINOR オプションは、HMINORオプションと同様で縦軸に関する指定である。

FOOTNOTE

*9 センター1階にある。

*10 TITLE作画位置は空白も含めて図の高さの中央となる。

```

DATA: INFILE DDD; INPUT X P;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=45 HPOS=90;
PROC GPLOT;
  PLOT P*X/VZERO VAXIS=1 10 100
      HZERO HAXIS=0 TO 2.5 BY 0.5 HMINOR=4
      VREF=2 4 6 8 10 20 40 60 80 100 LVREF=2
      HREF=0.5 TO 2.5 BY 0.5          LHREF=20;
  LABEL X=DISTANCE (M);
  TITLE1 .H=1.0 .F=SIMPLEX .A=90          PRESSURE (ATM);
  FOOTNOTE1 .H=2.0 .F=TRIPLEX PRESSURE DECAY WITH DISTANCE;
  FOOTNOTE2 .H=1.5 .F=COMPLEX (Fマオン=Pネナエトオ);

```

プログラム 5




図5 縦軸のラベルの作画方向の変更，対数軸の採用，目盛値間分割目盛線の指定
(縮尺0.7)

3.6 プロットシンボル種の指定、プロット点間を指定された線種の直線で結ぶ

SAS プログラムをプログラム 6 に、その出力結果を図 6 に示す。

```
DATA; INFILE DDD; INPUT X P;  
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20  
NOCHARACTERS VPOS=30 HPOS=60;  
PROC GPLOT;  
PLOT P*X/VZERO VAXIS=0 TO 30 BY 10 VMINOR=9  
HZERO HAXIS=0 TO 2.5 BY 0.5 HMINOR=4  
VREF=10 20 30 LVREF=27  
HREF=0.5 TO 2.5 BY 0.5 LHREF=21;  
LABEL X=DISTANCE (M);  
TITLE1 .H=1.0 .F=SIMPLEX .A=90 PRESSURE (ATM);  
SYMBOL V=DIAMOND L=1 I=JOIN;
```

プログラム 6




図 6 プロットシンボル種の指定、プロット点間を指定された線種の直線で結ぶ (縮尺 0.7)

【プログラム 6 の説明】

最下行のSYMBOLステートメント以外は、全てこれまでのプログラムと同じか、またはその変種である。

SYMBOLステートメントの説明

SYMBOLステートメントは、2つの機能を持つ。1つはプロットシンボルに関する指定を行うことであり、もう1つは、データ点間を結んだり補間したりする線に関する指定である。図6から図14までは、SYMBOLステートメントの種々のオプションの説明に充てられている。

ここで用いられているオプションの説明

V=記号名

プロットシンボルを指定する。記号とその名前との対応を図B（付録2）に示す。また記号名に **NONE** を指定するとプロットをしない。これらの他に文字をプロットシンボルとして用いることもできる。図9-2，図10，図15にその例を示す。図9-2はM、図15はo（小文字のO）を用いている。また、図10は文字として登録されている記号を用いている。

なお、SYMBOLステートメントを省略した場合には、図5に見られるようなシンボルが採用される（シンボル名はPLUSである。+ではないので注意）。また、SYMBOLステートメントを指定し、Vオプションを省略するとV=NONEが採用される。

Iの右に指定されるのは、データ点の連結法または回帰線に関する指定である。この指定の種類は多数あり、図14までに徐々に紹介する。ここの例で用いられている

I=JOIN

は、隣接する各点間を直線で結べという指定である。

またSYMBOLステートメントが省略された場合、または、Iオプションが省略された場合、I=NONEが採用され、どのような連結線も回帰線も作画されない。

L=線種のコード

Iオプションで指定した連結線または回帰線の作画に用いる線種をコード番号で指定する。線種とコード番号の対応を図A（付録1）に示す。Lオプションを省略した場合は、L=1が採用され実線で作画される。

3.7 プロット点間をスプライン補間して滑らかな曲線で結ぶ

SAS プログラムをプログラム7に、その出力結果を図7に示す。プログラムを簡単にするために、参照線等を作画しないことにした。

```
DATA; INFILE DDD; INPUT X P;  
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20  
NOCHARACTERS VPOS=30 HPOS=60;  
PROC GPLOT;  
PLOT P*X/VZERO VAXIS=0 TO 30 BY 10  
HZERO HAXIS=0 TO 2.5 BY 0.5;  
SYMBOL V=STAR I=SPLINE;
```

プログラム7




図7 プロット点間をスプライン補間して滑らかな曲線で結ぶ(縮尺0.7)

【プログラム 7 の説明】

SYMBOL ステートメントの I オプションのみを説明する。

I=SPLINE

と指定すると、各プロット点が滑らかな曲線で結ばれる。この際の補間法には 3 次スプライン関数^{*11} が用いられている。また両端のプロット点では曲率が 0 とされる。また、

I=SPLINES

と指定すると SAS はスプライン補間をする前に横軸の変数を並べ替えてくれる。

この指定は、これから説明する他の補間法についても有効である。即ち補間法を示すキーワード（例えば SPLINE）の語尾に S^{*12} をつければよい。

【プロット点を滑らかな曲線で結ぶその他の方法】

プロット点を滑らかな曲線で結ぶには、もう 1 つの方法がある。それはラグランジェ法を用いる方法である。その例をプログラム 7-2 と図 7-2 に示す。

I=L_n

がラグランジェ法を用いる場合の指定である。

n は 1, 3, 5 の値をとり、それに応じて n 次の多項式が補間式に用いられる。

これらの補間法の相違については文献 1 3 等を参照されたい。^{*13}

図 7、7-2 の例は一価関数であるが、SAS は多価関数のスプライン補間も行ってくれる。その例をプログラム 7-3 と図 7-3 に示す。

I=SPLINEP

が多価関数のスプライン補間を用いる場合の指定である。^{*14} もちろんプロット点は指定順に結ばれる。

同様に

I=L_n

と指定すると多価関数のラグランジェ補間をしてくれる。

FOOTNOTE

* 1 1 文献 1 3 等を参照せよ。

* 1 2 Sort (分類) の略と思われる。

* 1 3 最大の相違は、スプライン法は 2 次導関数も連続であるが、ラグランジェ法では連続とは限らないということである。しかし、見た目の滑らかさという点ではラグランジェ法の方がまさる。

* 1 4 語尾の P は Parametric の略と思われる。

```
DATA; INFILE DDD; INPUT X P;  
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20  
NOCHARACTERS VPOS=30 HPOS=60;  
PROC PLOT;  
PLOT P*X/VZERO VAXIS=0 TO 30 BY 10  
HZERO HAXIS=0 TO 2.5 BY 0.5;  
SYMBOL V=STAR I=L3;
```

プログラム7-2




図7-2 ラグランジェ法による補間(縮尺0.7)

```

DATA; INFILE DDD; INPUT X Y;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GLOT;
  PLOT Y*X/VZERO VAXIS=-7.5 TO 10.0 BY 2.5
       HZERO HAXIS=0 TO 11 BY 1;
  SYMBOL V=STAR I=SPLINEP;

```

プログラム7-3




図7-3 多価関数のスプライン補間(縮尺0.7)

3.8 プロット点間を階段状に結ぶ

SAS プログラムをプログラム8に、その出力結果を図8に示す。

【プログラム8の説明】

SYMBOLステートメントのIオプションのみを説明する。

I=STEPX₁X₂

と指定すると各プロット点をステップ的に結ぶ。

```

DATA; INFILE DDD; INPUT X P;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GPLOT;
  PLOT P*X/VZERO VAXIS=0 TO 30 BY 10
       HZERO HAXIS=0 TO 2.5 BY 0.5;
  SYMBOL V=TRIANGLE I=STEPCJ;

```

プログラム 8




図8 プロット点間を階段状に結ぶ(縮尺0.7)

X_1 をLとすれば、点がステップの左に、Rとすれば、右に、Cとすれば例のように中程にくる。

X_1 を省略すればLが採用される。

X_2 をJとすると、例のように各ステップが垂直線で結ばれる。 X_2 を省略すれば垂直線は作画されない。

3.9 プロット点の回帰線を引く

SAS プログラムをプログラム 9 に、その出力結果を図 9 に示す。

```
DATA; INFILE DDD; INPUT X P;  
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=30  
NOCHARACTERS VPOS=30 HPOS=60;  
PROC GPLOT;  
PLOT P*X/VZERO VAXIS=0 TO 30 BY 10  
HZERO HAXIS=0 TO 2.5 BY 0.5;  
SYMBOL V=SQUARE I=RC;
```

プログラム 9




図 9 プロット点の回帰線を引く (縮尺 0.7)

【プログラム 9 の説明】

SYMBOL ステートメントの I オプションのみを説明する。

`I=RX1X2X3n`

と指定すると^{*15}プロット点に回帰線を引いてくれる。X₃n については次節で説明することとして、ここでは X₁ と X₂ についてのみ説明する。

X₁ は回帰式の次数の指定であり、L, Q, C の 3 種類を指定することができる。それぞれ 1 次式 (Linear)、2 次式 (Quadratic)、3 次式 (Cubic) となる。

X₂ に 0 を指定すると回帰線は原点を通る。X₂ を省略すると原点を通るとは限らない。

【回帰線作画のための一般的方法】

上の例では、回帰式は 3 次までの多項式に限られるが次に、もっと一般的な形の回帰線を引く方法を示す。それには、非線形回帰モデルを分析するためのプロシジャ NLIN などの統計分析プロシジャと GPLOT を組み合わせて用いる。

その例をプログラム 9-2 と図 9-2 に示す。

【プログラム 9-2 の説明】

3 行目までは、これまでと同じ。

4 行目から 7 行目までは、NLIN プロシジャであり、詳細は、文献 3 の第 8 章を参照されたい。ここでは簡単な説明にとどめる。

このプロシジャでは X, P のデータに、MODEL で示される $P=1/(A \cdot X+B)$ の形の回帰式を当てはめ、係数 A, B を決定している。PARMS では A, B 決定作業のための初期値を与えている。そして、OUTPUT $_P=PP$ では、決定された A, B を用いて各 X に対する P の推定値を計算し、それを PP という変数名で記憶している。

従って、この NLIN プロシジャを終了した時点では、変数 X, P, PP が記憶されていることになる。これらのデータを用いて GPLOT プロシジャを実行する。

GPLOT プロシジャの PLOT ステートメントは、これまでとは若干違った形をしている。即ち次のようである。

`PLOT $_P$ *X $_PP$ *X/OVERLAY`

キーワード PLOT の後の P*X $_PP$ *X は、ここに指定した 2 組のプロットをせよという指定である。このような指定は幾つでも書くことができる。縦軸、横軸とも変数が異ってもよい。SAS が適当に軸を設定してくれる。次の OVERLAY オプションが無い場合は、別々の図を 2

FOOTNOTE

*15 筆者は最初の R は Regression (回帰) の略として記憶している。

```

DATA; INFILE DDD; INPUT X P;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC NLIN;
  PARS A=0.15 B=0.02;
  MODEL P=1/(A*X+B);
  OUTPUT P=PP;
PROC GLOT;
  PLOT P*X=1 PP*X=2/VZERO VAXIS=0 TO 30 BY 10
        OVERLAY HZERO HAXIS=0 TO 2.5 BY 0.5;
  SYMBOL1 V=M F=COMPLEX;
  SYMBOL2 V=NONE I=SPLINE;

```

プログラム 9-2




図 9-2 非線形回帰モデル分析による回帰線のプロット (縮尺 0.7)

枚作画する。但し、日本語端末などのディスプレイ上に出力する場合以外は、OVERLAY オプションは省略できない。OPR や NLP では1つのプロシジャで1枚の図しか作画できない。

PLOT P*X=1

縦軸変数×横軸変数= n と指定した n は、この変数の組の作画に用いるSYMBOLステートメントの番号を示している。一般には n は省略され、その場合、SYMBOLステートメントは番号順に用いられる。

OVERLAY

複数の変数の組のプロットを同じグラフ上に作画せよという指定である。

SYMBOLステートメントは2つ指定されている。特に指定しなければ最初の変数の組のプロットにはSYMBOL1ステートメントが、2番目の変数の組のプロットにはSYMBOL2ステートメントが用いられる。以下変数の組の数に従って幾つでも指定することができる。

次にSYMBOL1ステートメントに用いられているFオプションについて説明する。プロットシンボルには任意の文字を使うことができる。その際に次のように指定する。

F=文字の種類 V=文字 H=文字のサイズ

これらのオプションの意味はFOOTNOTEステートメントで説明した.F, .Hオプションと同じである。この例では、文字Mが文字の種類COMPLEXで作画されている。

3.10 プロット点の回帰線とその信頼区間線を引く

SAS プログラムをプログラム10に、その出力結果を図10に示す。

【プログラム10の説明】

SYMBOLステートメントのみを説明する。まずIオプションについて

I=RX₁X₂X₃n

このうち X_1X_2 については前節で説明した。この例では X_1 はCであり3次の回帰式を指示している。また X_2 は省略されており回帰線は原点を通るとは限らない。

X_3n は信頼区間線を書かせるための指定であり、 X_3 にはCLMまたはCLIを指定することができる。CLMと指定すれば平均予測値^{*16}の信頼区間を、CLIと指定すれば個々の予測値^{*16}の信頼区間を表わす2本の信頼区間線が回帰線をはさんで作画される。作画に用いられる線種は、回帰線の作画に用いられた線種コードに1を加えた線である。この例の場合はLオプション

FOOTNOTE

*16 詳細は文献3の第8章を参照されたい。

```

DATA; INFILE DDD; INPUT X P;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GPLOT;
      PLOT P*X/VZERO VAXIS=0 TO 30 BY 10
           HZERO HAXIS=0 TO 2.5 BY 0.5;
      SYMBOL V=i F=SPECIAL I=RCCLI;

```

プログラム 10




図10 回帰線とその信頼区間線を引く(縮尺0.7)

ンが省略されているので $L=1$ となり、回帰線は実線で作画され、信頼区間線は $L=2$ の破線で作画される。

n は、必ず X_3 と共に指定され、信頼区間線の信頼水準を指定する。指定可能な値は 90, 95, 99 のいずれかであり、省略すると 95 が採用される。なお、この例では、 $F=SPECIAL$ としたので、文字 I が \star になっている。特殊な字種の文字と記号の対応については文献 4 の付録 1 を参照されたい。^{*17}

3.11 パラメータ（第3変数）による複数のプロット

これまでプロットしてきたと同じような（X,P）のデータの組が、第3変数Qの値別に5組ある。これらのデータセットを、同一グラフ上にQの値によってプロットシンボルを変えてプロットする方法を示す。

SAS プログラムをプログラム11に、その出力結果を図11に示す。但し、データはX,P,Qの値がこの順に1行に1組ずつ書かれているものとする。

【プログラム11の説明】

新しく現われたステートメントやオプションは何もない。ただ PLOT ステートメントが若干異なるのみである。

PLOT 縦軸変数名*横軸変数名=パラメータ変数名

と指定すると、図3のようなプロットをパラメータ変数の値別に1つのグラフ上に重ねて行う。^{*18}
パラメータ変数の値は文字または文字列でも良い。但しパラメータ変数の値は16種類を超えてはならない。

プロットに際しては、プロットシンボルを勝手に変えてくれる。プロット点を連結したり回帰線を引いたり、プロットシンボルをユーザが指定する時は、例のように、Qの値の種類だけ SYMBOL ステートメントを用意する。Qの値の小さいものから順にSYMBOLステートメントが適用される。Qの値に文字を用いた場合も、文字に対応するEBCDICコードの小さいものの順となる。^{*19}

FOOTNOTE

*17 同じプロットシンボルが記号名と特殊文字名の2つの名前を持っているものがある。例えば、この図の特殊文字Iと図13の記号名=。そのような場合、プロットシンボルのサイズを特に指定しないと、特殊文字名で指定した場合は、基本文字サイズとなり、記号名で指定すると1/2サイズとなる。

*18 パラメータ変数の値別に別々のグラフとする方法は、プログラムD（付録4）に示す。

*19 アルファベット、数字の順となる。

```

DATA; INFILE DDD; INPUT X P Q;
      GOPTIONS DEVICE=SASPSY COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GLOT;
  PLOT P*X=Q/VZERO VAXIS=0 TO 30 BY 10
       HZERO HAXIS=0 TO 2.5 BY 0.5;
  SYMBOL1 V=_ L=2 I=JOIN;
  SYMBOL2 V=* L=4 I=JOIN;
  SYMBOL3 V=- L=8 I=JOIN;
  SYMBOL4 V+= L=10 I=JOIN;
  SYMBOL5 V=: L=15 I=JOIN;

```

プログラム 11




図11 パラメータ(第3変数)による複数のプロット(縮尺0.7)

3.12 繰返しのあるデータの最大値および最小値を示し、 各平均値を直線で結ぶ

この節では、図11に示したQ毎のプロットを、Qに関係のない、各X点での複数の測定値とみなす。そして図12のように、各X点でのデータの最大値と最小値を線で結び、即ちデータの存在範囲を示し、各平均値を直線で結ぶ。

SAS プログラムをプログラム12に、その出力結果を図12に示す。

【プログラム12の説明】

SYMBOLステートメントのIオプションのみを説明する。他はそれまで説明したものの組み合わせである。

`I=HILOX1X2`

と指定すると^{*20}、各X点でのデータの最大値と最小値を直線で結び、平均値に一印をつける。

X₁ は最大値と最小値の示し方の指示であり、TまたはBを指定する。^{*21} Tは例のように、最大値、最小値にも一印をつける指定である。Bは最大値、最小値を結ぶ線を2本にし棒状とする。

X₂ は各平均値を結ぶ指示であり、Jを指定することができる。そうすると例のように、各平均値が順に直線で結ばれる。但し、平均値を示す一印は作画されない。Jを指定しなければ各平均値は一印で示されるだけで相互に結ばれない。X₁とX₂はどちらを先に指定してもよい。

この他に更に語尾にCをつけてもよい。これはCLOSE値が見つかった時に一印をつけるという意味であるが、筆者には意味がわからない。

この例ではSYMBOLステートメントにIオプションしか指定していないが、当然他のオプションと組み合わせて用いて良い。

FOOTNOTE

*20 筆者はHILOはHigh Lowの略に違いないと思っている。

*21 Tは字体からの連想、BはBarの略と思われる。

```

DATA; INFILE DDD; INPUT X P Q;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GLOT;
      PLOT P*X/VZERO VAXIS=0 TO 30 BY 10
          HZERO HAXIS=0 TO 2.5 BY 0.5;
      SYMBOL I=HILOTJ;

```

プログラム 1 2




図1.2 繰返しのあるデータの最大値および最小値を示し、
各平均値を直線で結ぶ(縮尺0.7)

3.13 繰返しのあるデータの $3\sigma^{*22}$ を表示し、平均値を直線で結ぶ

前節と同様に、この節でも、図11に示したQ毎のプロットを、Qに関係のない、各X点での複数の測定値とみなす。そしてそれらのバラツキを計算して、その範囲を示し、各平均値を直線で結ぶ。

SAS プログラムをプログラム13に、その出力結果を図13に示す。

```

DATA; INFILE DDD; INPUT X P Q;
      COPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GPLOT;
  PLOT P*X/VZERO VAXIS=0 TO 30 BY 10
       HZERO HAXIS=0 TO 2.5 BY 0.5;
  SYMBOL V== H=0.7 I=STD3J;
  
```

プログラム13




図13 繰返しのあるデータの 3σ を表示し、平均値を直線で結ぶ(縮尺0.7)

【プログラム 13 の説明】

SYMBOL ステートメントの I オプションのみを説明する。他はこれまで説明したものの組み合わせである。

$I=STDnX_1X_2X_3X_4$ *23

n は表示するバラツキの範囲の指定であり、1, 2, 3 を指定することができる。それぞれ各 X 点の 1σ , 2σ , 3σ の範囲を直線で結んで表示する。省略すれば $n=2$ が採用される。

X_1 は σ の計算方法の指定である。省略すれば標準偏差となる。 X_1 に M を指定すると σ には標準偏差ではなく標準誤差*24 が計算される。

X_2 は σ の計算の基礎となる分散の計算方法の指定である。省略すれば通常の計算方法が用いられる。 X_2 に P を指定すると、分散は ANOVA モデル*25 と同様に計算される。

X_3 および X_4 は、前節の X_1 および X_2 と同じである。なお、 $X_1 \sim X_4$ は、どの順に指定されても良い。

ところで、この図で用いられているプロットシンボルは、図 10 に用いられているものと同じシンボルであるが、指定方法が異なる。図 10 では文字を用いているが、この図では記号名 '=' を用いている。これら 2 つは形は全く同じであるがサイズが異なる。特に指定しなければ、文字の場合は $H=1$ となり、記号の場合は $H=0.5$ となる。

また、このように文字にも同じ形のものがある記号に対しては、H オプションの指定が可能である。これに対し記号名 SQUARE などの対応した文字を持たない記号に H オプションを指定しても無視される。

FOOTNOTE

* 22 σ は標準偏差を意味する。

* 23 STD は S Tandard D eviation (標準偏差) の略と思われる。

* 24 σ を標準偏差、 n をデータ数とすると、標準誤差は σ / \sqrt{n} と表わせる。

* 25 文献 3 参照。各 X 点でのデータの繰返し数が等しくなければならない。

3.14 繰返しのあるデータのバラツキを考慮して、滑らかな相関線を引く

前節と同様に、この節でも、図11に示したQ毎のプロットを、Qに関係のない、各X点での複数の測定値とみなす。

```
DATA; INFILE DDD; INPUT X P Q;
      GOPTIONS DEVICE=SASPSY COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GPLOT;
  PLOT P*X/VZERO VAXIS=0 TO 30 BY 10
       HZERO HAXIS=0 TO 2.5 BY 0.5;
  SYMBOL V=# H=0.8 I=SM50S;
```

プログラム14




図14 繰返しのあるデータのバラツキを考慮して、滑らかな相関線を引く (縮尺0.7)

それらのバラツキを考慮して、必ずしも平均値を通らない適当な滑らかさの線を引く。

SAS プログラムをプログラム 14 に、その出力結果を図 14 に示す。

【プログラム 14 の説明】

SYMBOL ステートメントの I オプションのみを説明する。他はこれまで説明したものの組み合わせである。

`I=SMn` *26

n は 01 から 99 までの数字であり、数字が小さくなるほど、相関線は各データの平均値の近くを通る。逆に、この数字が大きくなると、相関線は各平均値を離れても滑らかさを優先するようになる。^{*27} 語尾 S は 3.7 で説明したように、横軸の変数を予め小さい順に並べかえてから補間するようにするための指定である。




図 14-2 I=SM10 と I=STD の OVERLAY 出力

FOOTNOTE

* 26 筆者は SMOOTH の略ではないかと思っている。

図14は $n=50$ としたものであるが、比較のために図14-2~4に $n=10, 50, 70$ の場合について示す。なお、これらの図では、データ点のプロットは省略し参考のために各データ点の 2σ と平均値が合わせて示されている。^{*28} なお、4.7で説明したのと同様に

$$I=SMnP$$

と指定すると、多価関数にも適用できる。




図14-3 I=SM50 とI=STDのOVERLAY 出力

FOOTNOTE

*27 ここでいう滑らかさとは、各区間にあてはめられたスプライン関数の2次導関数の2乗の積分の和で表わされるもので、相関線が1本の直線の場合が最も滑らかでその値は0となる。 n で指定しているのは、この値と、相関線からの残差の平方和との相対的重要度である。

*28 I=SMnオプションの作図と、前節で示したI=STDオプションの作図とをOVERLAYオプションを指定して重ね合わせたものである。




図14-4 I=SM70 とI=STDのOVERLAY 出力

3.15 指定した曲線間にハッチングをする

同一のグラフ上にプロットされた複数本の曲線によって囲まれた領域や、曲線と座標軸との間の領域に、適当な模様様のハッチングをすることができる。ここでは図10の信頼区間線の上に縦のハッチングをしてみる。

SAS プログラムをプログラム15に、その出力結果を図15に示す。

【プログラム15の説明】

新しく用いられているのは、PLOT ステートメントのAREASオプションとPATTERN ステートメントである。

AREAS=n

nは正の整数であり、下からn番目の領域までをハッチングの対象とすることを指示している。ここで下からn番目の領域とは、下からn-1番目の線とn番目の線との間の領域である。

```

DATA; INFILE DDD; INPUT X P;
      COPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GPLOT;
  PLOT P*X/VZERO VAXIS=0 TO 30 BY 10
      HZERO HAXIS=0 TO 2.5 BY 0.5
      AREAS=3;
  SYMBOL V=+ F=DUPLEX I=RCCLI;
  PATTERN1 V=E;
  PATTERN2 V=M1N90 R=2;

```

プログラム 15




図 15 指定した曲線間にハッチングをする (縮尺 0.7)

n が 1 の場合は X 軸と最下段の線の間領域となる。

この n の数に対応して、PATTERN ステートメントを n 個定義する。

PATTERN ステートメントの一般形は次の形である。

PATTERN n オプションの並び;

n の値の順に下側の領域から順に用いられる。

オプションには、V と R がある。

V= $X_1 n_1 X_2 n_2$

X_1 はハッチングをするか、しないか、または塗りつぶすかの指定であり、M, E, S^{*29} が指定できる。

E を指定すると、この例の最初の領域のようにハッチングをしない。

S を指定すると、その領域を塗りつぶすことができる。但し、NLP 出力の場合は白く抜けたようになってしまうので、あまり勧められない。

M を指定すると、この例の 2 番目の領域のように、その領域にハッチングする。

ハッチングのパターンは以下の $n_1 X_2 n_2$ で指定する。

V= $Mn_1 X_2 n_2$

n_1 はハッチングの濃淡の指定であり、1～5 の整数を指定することができる。値が大きくなれば濃くなる。

X_2 は、交差ハッチングをするかどうかの指定であり、X または N を指定する。

X を指定すると交差ハッチングを行う。

N を指定すると 1 方向の平行線群だけでハッチングをする。

n_2 はハッチング線の角度の指定であり、0～360 の値を指定することができる。省略すると 0 が採用される。また X_2 に X が指定されていると、 n_2 で指定した角度の線と直交する線も作図される。

R= n *30

PATTERN ステートメントの繰返し数を n で指定する。例えば、この例で最初の 2 領域を E とし 3 番目の領域だけにハッチングをするには、次のように指定してもよい。

PATTERN1 V=E R=2;

PATTERN2 V=M1N90;

2 番目の PATTERN ステートメントが 3 ではなく 2 となっているのに注意されたい。

FOOTNOTE

*29 M は Map の略。E は Empty の略。S は Solid の略と思われる。

*30 Repeat の略と思われる。

3.16 等高線図の作画


平面上の直方格子状にとられた点^{*31}の座標値とそれぞれの点での高さ（高さとして表現する第3変数の値）を与えると、SASはそれらの値から等高線を作画してくれる。

```

DATA;
  PA1=3.14159; ABN=6.02257E23; K=1.38053E-16; MOL=28;
  M=MOL/ABN;
  DO C=0 TO 3000 BY 100; DO T=300 TO 3000 BY 100;
    N=4*PA1*(M/2/PA1/K/T)**1.5*C*C*EXP(-M*C*C*5000/K/T)*1.E11;
  KEEP C T N; OUTPUT; END; END;
  GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
    NOCHARACTERS VPOS=45 HPOS=90;
PROC GCONTOUR; PLOT T*C=N;
  FOOTNOTE .H=2.0 .F=TITALIC MAXWELL DISTRIBUTION .H=1.5 (NITROGEN);

```

プログラム 16



MAXWELL DISTRIBUTION (NITROGEN)

図16 等高線図の作画（縮尺0.7）

ここでは温度 300K から3000Kにおける窒素分子の速度分布を、横軸に速度、縦軸に温度、高さ軸に分布確率をとって表わしてみる。データはマックスウェルの分布式に従って、DATA ステップで計算して与える。もちろん、これまでの例のように、計算はFORTRAN プログラムで行って、結果だけを読み込んでもよい。

SAS プログラムをプログラム 16 に、その出力結果を図 16 に示す。

【プログラム 16 の説明】

まず DATA ステップから説明する。^{*32}

1～6 行目までが DATA ステップである。FORTRAN プログラムと同様のプログラムが組まれているが、次の点には注意されたい。

DOステートメント : 機能は FORTRAN と同じであるが、形が少し異なっている。また、終りを示すのは END ステートメントである。

KEEPステートメント : SAS データとして記憶させるものだけを、この後に続けて並べる。これが無いと、SAS は、PAI や ABN などの値も、C, T, N と同様に、データの組数だけ毎回記憶することとなる。^{*33}

OUTPUTステートメント : SAS は一般には 1 回の DATA ステップの実行で 1 組のデータを記憶するだけである。従って、この例の場合は、DO ステートメントの最後の値だけしか記憶されないことになってしまう。それを避けるために用いられるのが、この OUTPUT ステートメントである。DATA ステップ中に何回現われてもよく、その度に、その時点での変数の値を記憶する。従って、DO ステートメントでデータを計算する場合には、この例のように使用されることが多い。

次に GCONTOUR プロシジャについて説明する。

9 行目だけが GCONTOUR プロシジャである。

PROC_GCONTOUR;

は、このプロシジャを始めるためのキーワードである。

PLOT_T*C=N; ^{*34}

は、T を縦軸に C を横軸に N を高さ軸にして等高線を作画せよという指定である。この後に、いくつかのオプションをつけることができるが、それは次節以降で説明する。

FOOTNOTE

* 3 1 全格子データがなくてもよい。50%以上のデータがあればよい。

* 3 2 詳細は、文献 2 または文献 6 を学習されたい。

* 3 3 別に邪魔にはならないが、それだけ容量や時間を余分に必要とする。

なお、この例のように、縦軸と横軸の単位が異っているものは良いが、地図などのように縦軸と横軸の単位が同じでしかもサイズを合わせたい場合には、SAS による作画はあまり適当でない。既に述べたように SAS による図の縦軸と横軸の長さを意図どおりに調整するのは、非常に困難なことである。指定できるのは、座標軸等を含めた図全体の縦と横のサイズである。

3 . 1 7 等高水準値とその使用線種の指定

図16では等高水準値の設定を SAS にまかせたため、39や68など区切りの悪い値の等高線になってしまった。^{*35}

そこで、ここでは等高水準値を指定する。ついでに等高線種も指定する。

SAS プログラムをプログラム17に、その出力結果を図17に示す。

【プログラム17の説明】

このプログラムの DATA ステップでは、データの SAS データセットへの登録を行っている。

DATA ステートメントの後ろにオプションとして

ファイル名.メンバ名

と指定すれば、そのファイル名の所に予め割り付けられた^{*36} SAS データセットに、指定したメンバ名でデータが登録される。

この SAS データセットの取り扱いは、通常のデータセットとは異なるので注意を要する。利用法については次節で紹介する。詳細は文献2の第5章を参照されたい。

GCONTOUR プロシジャは、ほとんどプログラム16と同じであり PLOT ステートメントにオプションがついただけである。

FOOTNOTE

*34 同じ形の PLOT ステートメントが GPLOT プロシジャ内で用いられた場合には、意味が異なる(3.11参照)。

*35 等高水準値を特に指定しなければ、SAS は最小値と最大値の間を5%から95%まで15%間隔に区切った7水準を採用する。

*36 通常の SAS プログラムの実行と同様に F()とDA()で指定すればよい。


INFILE ステートメントも使っていて、2つのファイルを使用する必要がある時は、F(YES)と指定しDA()は指定しない。そうすると、ファイル名とデータセット名を順に尋ねて来るので、それに答えればよい。

```

DATA DDD.MAXWELL;
  PAI=3.14159; ABN=6.02237E23; K=1.38053E-16; MOL=28;
  M=MOL/ABN;
  DO C=0 TO 3000 BY 100; DO T=300 TO 3000 BY 100;
    N=4*PAI*(M/2/PAI/K/T)**1.5*C*C*EXP(-M*C*C*5000/K/T)*1.E11;
  KEEP C T N; OUTPUT; END; END;
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
          NOCHARACTERS VPOS=45 HPOS=90;
PROC GCONTOUR;
  PLOT T*C=N/LEVELS=5 10 20 TO 160 BY 20
        LLEVELS=22 21 20 10 9 8 4 3 2 1;
  FOOTNOTE .H=2.0 .F=TITALIC MAXWELL DISTRIBUTION .H=
          1.5 .F=DUPLEX(NITROGEN);

```

プログラム 17



MAXWELL DISTRIBUTION (NITROGEN)

図17 等高水準値とその使用線種の指定 (縮尺0.7)

LEVELS=等高水準値の並び

指定した等高水準値の等高線が作画される。

等高水準値の指定に際しては、数値を順に並べても良いし、DO形式を用いても良い。また両法を並用しても良い。作画できる等高線の最大数は100本である。

LLEVELS=線種コードの並び

等高線の作画に用いる線種をコード番号で指定する。

等高水準値の指定数と同数だけ指定しなければならない。省略すれば、1番から32番までの線種が順に使用される。

3.18 ハッチングパターンによる等高水準表示

図18のように、等高水準をハッチングパターンで示すこともできる。但し、表示単位はX-Y座標の目盛線単位に分割された長方形である。従って位置表示は粗くなる。

SAS プログラムをプログラム18に、その出力結果を図18に示す。

【プログラム18の説明】

このプログラムでは、前節でSASデータセットにメンバ名MAXWELLで登録したデータを用いている。

SASデータセット内のデータを用いるのは簡単で、DATAステップの先頭で

SET_ファイル名.メンバ名

とすればよい。但し、ここで指定されたファイル名の所には、このメンバの存在するSASデータセットが予め割りつけてなければならない。

PLOT ステートメントのオプション

PATTERN は、ハッチング・パターンによる等高表示をせよという指定である。なお、ハッチングに使用されるパターンは、別にPATTERNステートメントによって指定される。

JOIN は、PATTERNオプションと共に用いられ、この指定を省略すると、各格子の区切り線が全て作画される。

PATTERNステートメント は、当然、等高水準の数だけ必要である。また、このステートメントはグローバル・ステートメントであり、DATAステップに置いても良い。このステートメントおよびオプションについては、既に3.15で述べた。こちらの方が実例が豊富なので、Vオプションの指定と凡例の模様を比較検討されたい。

```

DATA; SET DDD.MAXWELL;
      COPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
      NOCHARACTERS VPOS=45 HPOS=90;
PROC GCONTOUR;
  PLOT T*C=N/LEVELS=5 10 20 TO 160 BY 20 PATTERN JOIN;
  PATTERN1 V=E;
  PATTERN2 V=M1;
  PATTERN3 V=M1N90;
  PATTERN4 V=M1X45;
  PATTERN5 V=M3;
  PATTERN6 V=M3N90;
  PATTERN7 V=M3X45;
  PATTERN8 V=M5;
  PATTERN9 V=M5N90;
  PATTERN10 V=M5X45;

```

プログラム 18




図18 ハッチングパターンによる等高表示 (縮尺0.7)

3.19 鳥瞰図の作画

等高線を作画したのと同様のデータ（直方格子上の第3変数）^{*37}を用いて、SASは3次元立体図（いわゆる鳥瞰図）を作画することができる。

そのSASプログラムをプログラム19に、作画結果を図19に示す。

```
DATA; SET DDD.MAXWELL;  
  GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20  
  NOCHARACTERS VPOS=45 HPOS=90;  
PROC G3D;  
  PLOT T*C=N;  
  FOOTNOTE1 .H=2.0 .F=TITALIC MAXWELL DISTRIBUTION;  
  FOOTNOTE2 .H=1.2 .F=ITALIC (NITROGEN);
```

プログラム19




図19 3次元鳥瞰図の作画（縮尺0.7）

FOOTNOTE

* 37 全格子点のデータが無くても、50%以上のデータがあればよい。

【プログラム 19 の説明】

DATA ステップは、プログラム 18 と全く同じである。

4 行目と 5 行目で鳥瞰図を作画している。

```
PROC_G3D;
```

鳥瞰図作画のためのプロシジャ名 G3D の宣言である。

```
PLOT_T*C=N;
```

T を縦軸 (Y 軸) に、C を横軸 (X 軸) に、N を高さ軸 (Z 軸) にとって鳥瞰図を作画するよう指示している。一般形は、

```
PLOT_Y軸変数名*X軸変数名=Z軸変数名/オプションの並び;
```

である。オプションについては、次節以降で紹介する。

3 . 2 0 鳥 瞰 図 の 回 転 (Z 軸 回 り)

前節で作画した立体を回転させて見易い角度を探してみよう。この節では Z 軸回り (図 19 では N 軸)、次節で Y 軸回り (図 19 では T 軸)、さらに次次節で、これらの組合せについて紹介する。なお、X 軸回りに回転させる方法はない。

SAS プログラムをプログラム 20 に、その作画結果を図 20 に示す。図 19 と比較して、回転の軸と方向を確認していただきたい。

【プログラム 20 の説明】

プログラム 19 から FOOTNOTE が除かれている。

それ以外の相違は、PLOT ステートメントのオプションのみである。

```
ROTATE=角度
```

で Z 軸回りの回転角を指定する。この指定を省略すると、ROTATE=70 が採用される。図 19 はこの状態である。このプログラムでは ROTATE=30 と指定しているので、図 20 は図 19 から負の方向に 50° 回転したものとなっている。従って極端な場合として角度に 0 を指定すると Y 軸は Z 軸と同一直線上になり、X 軸は Z 軸に垂直となる。

なお、角度は 0~90 まで指定できる。

また

```
ROTATE=角度の並び *38
```

とすると、指定された角度毎に 1 枚ずつ作図する。

なお、この指定法はディスプレイ画面上でしか用いることができない。理由は 3 . 9 で述べた。


```

DATA; SET DDD.MAXWELL;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
      NOCHARACTERS VPOS=45 HPOS=90;
PROC G3D;
      PLOT T*C=N/ROTATE=30;

```

プログラム 2 0




図 2.0 鳥瞰図の回転 (Z 軸回り) (縮尺 0.7)

FOOTNOTE

* 3 8 DO 形式で指定しても良い。

3.21 鳥瞰図の回転（Y軸回り）

SAS プログラムをプログラム21に、その作画結果を図21に示す。図19と比較して、回転の軸と方向を確認していただきたい。

```
DATA; SET DDD.MAXWELL;  
  GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20  
  NOCHARACTERS VPOS=45 HPOS=90;  
PROC G3D;  
  PLOT T*C=N/TILT=50;
```

プログラム21




図21 鳥瞰図の回転（Y軸回り）（縮尺0.7）

【プログラム21の説明】

プログラム20とほぼ同じであり、相違はROTATEオプションの代わりに、次のオプションを使用したのみである。

TILT=角度

Y 軸回り (図 2 1 では T 軸) の回転角の指定であり、省略するとこれも 70° が採用される。
極端な場合として角度 90 を指定すると、X-Y 平面が一本の直線になってしまう。

この角度の指定の範囲も 0~90 である。

また、ROTATE と同様に

TILT=角度の並び

とも指定できる。その指定に関する注意も ROTATE の場合と同じである。

3. 2. 2 目盛値作画様式の指定、参照線の指定

図 2 0 や図 2 1 のままでは、回転角によっては、目盛値が軸と交差してしまうことがある (図 2 0 がその状態に近い)。ここでは、それを避けると同時に、ちょっとカッコをつけて目盛値を作画してみる。

また、各軸に対する参照線の作画法についても紹介する。

SAS プログラムをプログラム 2 2 に、その作画結果を図 2 2 に示す。

【プログラム 2 2 の説明】

PLOT ステートメントのオプションのみを説明する。

FANCY

このオプションを指定すると、目盛値およびラベルも適当な平面上に書かれてあるように作画される。これ以上に特別な指定をしなければ、図 2 2 の例のように、X 軸目盛値は X-Z 平面上、Y 軸目盛値は Y-Z 平面上に、Z 軸目盛値も Y-Z 平面上にあるように作画される。

また、FANCY オプションを指定し、さらに FLAT オプションも指定すると、次節の図 2 3 のように、X 軸目盛値および Y 軸目盛値は X-Y 平面上にあるように作画される。

GRID

各軸の目盛値から参照線を引く。但し、図 2 2 に示すように、X, Y 軸に関する参照線は X-Y 平面上にしか作画されない。

鳥瞰図では、特定の軸に関してのみ参照線を引くことはできない。また、参照線の値をユーザが指定することはできない。

LABSIZE=文字サイズ

各軸の目盛値およびラベル作画に用いる文字のサイズを指定する。単位は基本文字サイズである。

```

DATA; SET DDD.MAXWELL;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=45 HPOS=90;
PROC G3D;
      PLOT T*C=N/ROTATE=30 TILT=80 FANCY GRID LABSIZE=1.5;

```

プログラム 2 2




図 2 2 目盛値作画様式の指定、リファレンス線の指定 (縮尺 0.7)

3 . 2 3 鳥瞰図を一方向だけの線で作画する

図 2 2 の鳥瞰図は X-Y 両方向の曲線で作画しているが、これが煩わし過ぎる場合がある。どちらか一方の曲線に主に注目している時などである。そのような場合は一方向だけの線で鳥瞰図を作画することができる。

SAS プログラムをプログラム 2 3 に、その作画結果を図 2 3 に示す。

```

DATA; SET DDD.MAXWELL;
      GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
              NOCHARACTERS VPOS=45 HPOS=90;
PROC G3D;
      PLOT T*C=N/ROTATE=30 TILT=80 XYTYPE=1 LABSIZE=1.5 GRID
              FANCY FLAT;

```

プログラム 23




図 23 鳥瞰図を一方向だけの線で作画する (縮尺 0.7)

【プログラム 23 の説明】

PLOT ステートメントのオプションのみ説明する。

XYTYPE=n

n は 1, 2, 3 のどれかであり、どの方向の曲線で鳥瞰図を作画するか指定である。

1 は X 方向の線のみ、2 は Y 方向の線のみ、3 は両方向の線を作画する。このオプションを指定しなければ、3 が採用される。

FANCY FLAT

4.22 で説明済。

3.24 非直方格子データによる等高線図および鳥瞰図の作画

3.16～3.23に示した等高線や鳥瞰図の例では、そのX-Yデータが直方格子状に取られたものである必要があった。ここでは、そのような制約を受けない非直方格子状のデータから等高線および鳥瞰図を作画する方法を示す。極座標系のデータや有限要素法による計算等の作画に有効であろう。

そのようなデータは、1つの前処理をして直方格子データに直し、その後これまでの処理をする。この前処理もSASのプログラム内で行う。

SASプログラムをプログラム24に、作画結果を図24-1、図24-2に示す。

```
DATA;
  PA1=3.14159;
  DO R=1.0 TO 15.0 BY 1.0; DO T=0.0 TO PA1/2 BY PA1/20;
    X=R*SIN(T); Y=R*COS(T); Z=(COS(8*T))**2/R;
  OUTPUT; END; END;
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
          NOCHARACTERS VPOS=30 HPOS=60;
PROC G3GRID; GRID Y*X=Z/SPLINE NAXIS1=21 NAXIS2=21;
PROC G3D; PLOT Y*X=Z/ROTATE=35 TILT=75 FANCY;
PROC GCONTOUR;
  PLOT T*X=Z/LEVELS=-0.7 TO 2.5 BY 0.2;
```

プログラム24

【プログラム24の説明】

5行目までは、DATAステップである。極座標を用いているので、X-Yデータは格子状になっていない。詳細な説明は省略する。

8行目のG3GRIDプロシジャが、非直方格子データを補間して直方格子データに変換するプロシジャである。

PROC G3GRID;

は、キーワードであり、通常このまま書く。

GRID 縦軸変数名*横軸変数名=高さ軸変数名/オプションの並び;

GRID ステートメントは、データの補間に関する指示を与える。

ここで用いられているオプションの意味は次のようである。




図24-1 非直方格子データによる鳥瞰図の作画(縮尺0.7)




図24-2 非直方格子データによる等高線図の作画(縮尺0.7)

SPLINE

補間に際して2変数スプライン法を用いることを指示する。省略すると標準法^{*39}が用いられる。

NAXIS1=格子線数

補間して求めるべき縦軸変数値の数を指定する。即ち、分割数+1となる。省略すれば、11が採用される。

また、補間値を求めるべき変数値を直接指定することもできる。その場合は、次のようにする。

AXIS1=値のリスト (昇順)

値のリストは必ず昇順に並べる。

なお、NAXIS1オプションと同時に指定された場合は、AXIS1オプションが優先される。

NAXIS2=格子線数

AXIS2=値のリスト (昇順)

横軸に変する指定である。意味はNAXIS1, AXIS1と同様である。

9、10行目はそれぞれG3D、GCONTOURプロシジャである。意味は、既に説明済である。このように、1つのプログラムで幾つものプロシジャを並べることができる。図は、もちろんプロシジャ別に別図として出力される。この方法は、OPRやNLPへ出力する場合も指定可能である。

なお、格子間隔の粗いデータをそのまま3次元鳥瞰図にすると、プロシジャG3Dは何ら補間をせずにデータ点間を結ぶだけなので、きれいな図にならないことがある。そのような時に、この前処理プロシジャG3DGRIDによって細かく補間してやれば、滑らかな図を作ることができる。

【サイズの異なる2枚の図の作画】

ところで、図24-1, 図24-2は、1つのSASプログラムの実行によって、一度に出力されたものであるが、図24-2は、横長になり過ぎて見にくい図となっている。そこで次に、1つのSASプログラムの実行の途中で、図のサイズを変更してみよう。

SASプログラムを、プログラム24-3に、作画結果を図24-3に示す。鳥瞰図の方は、図24-1と全く同じ図となるので、このプログラムの作画結果からは省略した。

FOOTNOTE

*39 詳細は文献4のG3GRIDプロシジャを参照されたい。標準法は、あまり上手な補間法ではない。時間はかかるが(従って料金も)スプライン法を用いた方がきれいである。

なお、標準法とスプライン法の間違った方法があり、その場合はSPLINEの代りにPARTIALと指定する。


```

DATA;
  PAI=3.14159;
  DO R=1.0 TO 15.0 BY 1.0; DO T=0.0 TO PAI/2 BY PAI/20;
    X=R*SIN(T); Y=R*COS(T); Z=(COS(8*T))**2/R;
  OUTPUT; END; END;
PROC G3GRID; GRID Y*X=Z/SPLINE NAXIS1=21 NAXIS2=21;
PROC G3D; PLOT Y*X=Z/ROTATE=35 TILT=75 FANCY;
GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
          NOCHARACTERS VPOS=30 HPOS=60;
PROC GCONTOUR;
  PLOT Y*X=Z/LEVELS=-0.7 TO 2.5 BY 0.2;
GOPTIONS VSIZE=15 HSIZE=12 VPOS=45 HPOS=55;

```

プログラム 24-3




図 24-3

【プログラム 2 4 - 3 の説明】

GOPTIONS ステートメントが2回指定されている。これまでの指定位置と異っているので注意されたい。複数の PROC を持つプログラムを作る時には、このプログラムの位置に置く方が良くであろう。

一般に、ある PROC ステップに適用されるグローバル・ステートメントは、プログラムの先頭から次の PROC ステップの前までに指定されたもののうちの最新のものである。従って、GOPTIONS ステートメントも SYMBOL ステートメントなどと同様に、適用すべき PROC ステップの直後に指定した方が間違いが少ないであろう。

なお、2回目以降は変更するオプションについてのみ指定すればよい。この例では、図のサイズと文字サイズの変更をしている。等高線図は正方形としたいのであるが、指定できる図のサイズはグラフ部分のみのサイズではなく、凡例や軸目盛なども含んだサイズであるので、グラフ部分のみを正確に正方形とすることは非常に困難である。

3 . 2 5 文字および記号のみの作画

SAS プログラムをプログラム 2 5 に、その作画結果を図 2 5 に示す。

【プログラム 2 5 の説明】

文字を作画するだけなので、DATA ステップはない。

GOPTIONS ステートメントに1つ新しいオプションを加えた。

BORDER

図全体の枠を作画せよという指定である。もちろん通常のグラフの作画にも適用できる。

3行目以降が全て、GSLIDEプロシジャである。

PROC GSLIDE ;

文字や記号の作画のみを行うためのプロシジャの宣言である。どちらもキーワードであり、通常はこのまま指定する。

GSLIDEプロシジャで用いることができるステートメントは、TITLE, FOOTNOTE, NOTE の3種類である。これらの3ステートメントは指定可能なオプションも全て共通である。FOOTNOTE ステートメントについては、既に4.4で述べたので、他の2つの違いをそれとの比較で説明する。

TITLE n \square オプション \square テキスト ;

FOOTNOTE が図の下側に文字を作画するのに対して、TITLEは図の上側に作図する。また、TITLE1に限り、文字サイズの指定を省略すると .H=2 が採用される。その他に関して

```

GOPTIONS DEVICE=SASPSP COLORS=(BLACK) VSIZE=15 HSIZE=20
          BORDER NOCHARACTERS VPOS=15 HPOS=20;
PROC GSLIDE;
  TITLE1 .H=0.5 .J=LEFT .F=SPECIAL ABCDEFG .H=
        0.7 .J=CENTER .F=CSCRIPT SAS/GRAPH .H=
        0.5 .J=RIGHT .F=SPECIAL GFEDCBA;
  TITLE3 .H=2.0 .F=OLDENG S A S;
  TITLE4 .H=0.3;
  TITLE5 .H=1.0 .F=TITALIC S.F=ITALIC マネオネ .F=
        TITALIC A.F=ITALIC ニツクウアホナト .F=
  TITLE6          TITALIC S.F=ITALIC モヘオチ;
  TITLE7 .H=0.7 .F=HIRA FVY;
  TITLE9 .H=1.0 .F=KAN4 D.F=KAN3 E.H=
        0.7 .F=KATA 5VK~ .H=1.0 .F=KAN6 '' .F=KAN2 ァ.H=
        0.7 .F=KATA 1 .H=1.0 .F=KAN1 %P;
  FOOTNOTE1 .H=1.0 .F=TRIPLEX SOME EXAMPLES;
  FOOTNOTE2 .H=0.5;
  FOOTNOTE3 .H=0.7 .F=COMPLEX X.M=(+0,+0.4) .H=0.4 2;
  FOOTNOTE4 .H=0.3;
  FOOTNOTE5 .H=0.7 .F=CGREEK DGLPQSWY-.F=CYRILLIC BCDHILQY .D=
        (+0,-1,+0,+2,-9.5,+0,+0,-2,+9.5,+0);
  TITLE10 .H=0.7 .F=DUPLEX .A=90 TITLE10;
  FOOTNOTE6 .H=0.7 .F=SIMPLEX .A=90 FOOTNOTE6;
  FOOTNOTE7 .H=0.7 .F=KAN6 .A=-90 .R=90 Oウ.F=HIRA 7;
  NOTE .H=0.5 .F=DUPLEX KANAZAWA UNIVERSITY;
  NOTE .H=0.6 .F=COMPLEX Y.NAKANO;

```

プログラム 25

は、FOOTNOTE と同じである。

NOTE_オプション_テキスト;

FOOTNOTE やTITLEがその後尾に数字を指定できるのに対して、NOTE は指定できない。プログラムでの出現順に作画される。そのかわり、ステートメントの数に制限はない。

また、FOOTNOTE やTITLEは特に指定しなければ、中央揃えになるのに対して、NOTE は特に指定しなければ左詰めとなる。

その他に関しては、FOOTNOTE と同じである。

次に、このプログラムで新しく用いられたオプションについて説明する。

.J=X *40

文字列の作画位置を指定する。

このオプションは、TITLE1 で用いられている。

Xに LEFT (短縮形L) を指定すると左側に詰めて作画される。




図 2 5 文字および記号のみの作画 (縮尺 0.7)

XにCENTER (短縮形C) を指定すると中央に揃えて作画される。

XにRIGHT (短縮形R) を指定すると右側に詰めて作画される。

なお、この指定を省略すると、FOOTNOTEとTITLEではCが採用され、NOTEではLが採用される。

`.M=(x,y)` *41

直前の文字作画終了位置からの相対的移動量を指定して、次の文字作画開始位置を指定することができる。このオプションは、FOOTNOTE3で用いられ、 X^2 の上付文字2を作画している。x,yはそれぞれX方向、Y方向の相対的移動量の指定であり、必ず+または-の記号をつけて指定する。単位は基本文字サイズである。

FOOTNOTE

* 4 0 JはJustification (整頓)の略と思われる。

* 4 1 MはMove (移動)の略と思われる。

.D=($x_1, y_1, x_2, y_2, \dots, x_n, y_n$) *42

(x_i, y_i) で示される座標を順に直線で結ぶ。

ここでは、FOOTNOTE5 で用いられ、最下行のギリシャ文字およびキリール文字の枠を作画している。

x_i, y_i の意味については、M オプションと同じである。 x_2, y_2 以降もすべて直前の作画位置からの相対移動量である。

.A=角度 *43

文字列の作画方向の回転角を指定する。指定可能な角度範囲は $-90^\circ \sim 90^\circ$ である。省略すれば当然 0° が採用される。

なお、角度に 90 または -90 を指定した場合に限り、作画位置も移動する。即ち

.A=-90 の場合、TITLE は右端に、FOOTNOTE は左端に作画される。ここでは FOOTNOTE7 で用いられ、図の左端に“縦書き”と書いている。

.A=90 の場合、TITLE は左端に、FOOTNOTE は右端に作画される。ここでは TITLE10 と FOOTNOTE6 で用いられ、それぞれ図の左端および右端に文字を作画している。

.R=角度 *44

各文字の回転角を指定する。指定可能な角度は $0^\circ \sim 360^\circ$ である。省略すれば **.R=0** (文字列作画方向に対して直立) が採用される。

ここでは、FOOTNOTE7 で、**.A** オプションと共に用いられ、“縦書き”という文字が横倒しに書かれるのを修正している。

なお、TITLE4, FOOTNOTE2 等は単に行間隔の調整のみのために用いられている。

なお、TITLE5 等のように、文字列の途中でオプションを変更する時には、余分な空白を入れないように注意しなければならない。また、ステートメントの途中で改行する場合も同様である。

FOOTNOTE ステートメント等で空白が無視されるのは、次の場合だけである。

- ・キーワード FOOTNOTE, TITLE, NOTE の直後に続く空白。
- ・=記号の後およびオプション値の前に続く空白。
- ・オプション値の直後の空白1つだけ。

FOOTNOTE


* 4 2 Dは Draw (線を引く) の略と思われる。

* 4 3 AはAngle (角度) の略と思われる。

* 4 4 RはRotation (回転) の略と思われる。

付録1 SASで使える線種とそのコード番号

SASで使える線種は32種であり、それらの線種とコード番号を図Aに示す。縦軸の目盛として示されているのがコード番号である。またX=5の参照線にはコード番号32の線種が用いられている。なお、ユーザがピッチの指定をすることはできない。



図A SASで使える線種とそのコード

付録2 SASで利用できるプロットシンボルとそのシンボル名

SAS で利用できるプロットシンボルは31種であり、それらのプロットシンボルとシンボル名を図Bに示す。3個続けて横に並んでいるのがプロットシンボルであり、その右側に書かれているのがシンボル名あるいはシンボル記号である。

なお、左上隅のシンボル記号は. (ピリオド) であるが、図に出力することはできなかった。また、文字のうちのいくつかの字種には記号が対応しており、それらもプロットシンボルとして利用することができる。詳細は文献4)を参照されたい。

?	?	?		h	h	h	<
<	<	<	[⊕	⊕	⊕	+
⊗	⊗	⊗	&	◇	◇	◇	羊
♀	♀	♀	×	♂	♂	♂)
○	○	○	-	♀	♀	♀	/
⊕	⊕	⊕	%	□	□	□	—
♂	♂	♂	>	B	B	B	?
*	*	*	:	♥	♥	♥	#
♂	♂	♂	⊙	♂	♂	♂	'
*	*	*	=	◇	◇	◇	"
◇	◇	◇	DIAMOND	#	#	#	HASH
⋯	⋯	⋯	PAW	+	+	+	PLUS
□	□	□	SQUARE	*	*	*	STAR
△	△	△	TRIANGLE	X	X	X	X
Y	Y	Y	Y	Z	Z	Z	Z

図B SASで利用できるプロットシンボルとそのシンボル名

付録3 SASで利用できる字種とその例

SASで利用できる字種には、図Cに示したものの他に、プロット用の記号や数字記号、音楽記号などがある。また、ひらがな、カタカナおよびいくつかの漢字もある。詳細は文献4)を参照されたい。図Cの左端に書かれたものが字種名であり、字体例はそれぞれ左から順に対応している。

SIMPLEX	ABCDEFGH	stuvwxyz
DUPLEX	ABCDEFGH	stuvwxyz
COMPLEX	ABCDEFGH	stuvwxyz
TRIPLEX	ABCDEFGH	stuvwxyz
SCRIPT	<i>ABCDEFGH</i>	<i>stuvwxyz</i>
CSCRIPT	<i>ABCDEFGH</i>	<i>stuvwxyz</i>
ITALIC	<i>ABCDEFGH</i>	<i>stuvwxyz</i>
TITALIC	<i>ABCDEFGH</i>	<i>stuvwxyz</i>
GITALIC	<i>ABCDEFGHI</i>	<i>stuvwxyz</i>
OLDENG	A B C D E F G H	stuvwxyz
GERMAN	A B C D E F G H	stuvwxyz
GREEK	ΑΒΓΔΕΦΓΗ	στυθωχψξ
CGREEK	ΑΒΓΔΕΦΓΗ	στυθωχψξ
CYRILLIC	АБЦДЕФГЖ	СТУВЩХЫІЗ

図C SASで利用できる主な字種とその例

付録4 パラメータの値毎の同一処理

3.11で示したように、パラメータQの値別に(X,P)のデータの組が多数ある場合、3.11ではX-PのプロットをQの値別に同一グラフ上に重ねて行った。


ここでは、Qの値別に別々のグラフとしてみる。

SAS プログラムをプログラムDに、その出力結果のうち1つだけを図Dに示す。図Dのような図がQの値の数だけ出力される。

```
DATA; INFILE DDD; INPUT X P Q;  
GOPTIONS DEVICE=F6653 VSIZE=7 HSIZE=10  
NOCHARACTERS VPOS=30 HPOS=60;  
PROC SORT; BY Q;  
PROC GPLOT UNIFORM; BY Q; PLOT P*X;  
SYMBOL V=+ L=1 I=JOIN;
```

プログラムD

Q=0.67



図D プログラムDの出力結果の一部

【プログラム D の説明】

これまでと変わったのは4行目と5行目である。まず SORT プロシジャについて説明する。

PROC_SORT;

は、SORT プロシジャを始めるためのキーワードである。このプロシジャは次の BY ステートメントで指定された変数の値の大小によって、データを並べ替えてくれる。

BY_変数名の並び;

SORT プロシジャ内に用いられる BY ステートメントは、並べ替えの変数名を指定する。

特に指定しなければ昇順に（小さいものから順に）並べ替えられる。降順に並べ替えるには、変数名の前に DESCENDING と指定する。例えば

```
BY _DESCENDING_Q
```

また、複数の変数名を指定すると、まず最初の変数値によって並べ替え、次に並べ替えられたそれぞれのグループの中で、第2の変数値によって並べ替えられる。後は、同様である。

次に GPLLOT プロシジャの新しい部分について説明する。

PROC_GPLOTTUNIFORM;

UNIFORM オプションを指定すると、このプロシジャで作画される、すべてのグラフが全く同じ軸となる。このオプションを指定しないと、それぞれのグラフはそれぞれのデータによって適当な軸を設定される。

なお、軸目盛があらかじめ設定できる場合には、このオプションを指定しないで、PLOT ステートメントで VAXIS, HAXIS オプションを指定しても、すべてのグラフの軸が同じとなる。

BY_変数名;

GPLLOT プロシジャ内に用いられる BY ステートメントは、このステートメントで指定した変数の値別に、処理を行うことを指示している。但し、ここで指定した変数名によって、あらかじめ SORT プロシジャによる並べ替えが行われていなければならない。

この例では、Q の値別に X-P プロットが行われている。なお、図 D に示すように、グラフ毎にグラフの上部に Q の値が表示される。

なお、BY ステートメントによる複数のグラフの作画は、OPR や NLP への出力の場合には、用いることはできない。

付録5 カラー機能を用いたプログラム例

プログラム例をプログラムE, F, G, Hに示す。すべてのプログラムのDEVICE指定がF6653となっていることに注意されたい。

プログラムEはプログラム11の変形である。CVREF, CHREFは参照線の色指定であり、CAXISは軸の色指定であり、CTEXTは目盛値、ラベルの色指定である。また、TITLE, FOOTNOTE およびSYMBOLステートメントの.Cオプションは、それぞれの色の指定である。

```
DATA; INFILE DDI; INPUT X P Q;
      GOPTIONS DEVICE=F6653 VSIZE=7 HSIZE=10
              NOCHARACTERS VPOS=30 HPOS=60;
PROC GPLOT;
      PLOT P*X=Q/VZRO VAXIS=0 TO 30 BY 10
           HZERO HAXIS=0 TO 2.5 BY 0.5
           VREF=10 20 30           LVREF=27 CVREF=GREEN
           HREF=0.5 TO 2.5 BY 0.5 LHREF=21 CHREF=PINK
           CAXIS=CYAN CTEXT=RED ;
      SYMBOL1 C=YELLOW V= L=2 I=JOIN;
      SYMBOL2 C=BLUE V=¥ L=4 I=JOIN;
      SYMBOL3 C=GREEN V=- L=8 I=JOIN;
      SYMBOL4 C=RED V=+ L=10 I=JOIN;
      SYMBOL5 C=CYAN V=: L=15 I=JOIN;
```

プログラムE

```
DATA; SET DDD.MAXWELL;
      GOPTIONS DEVICE=F6653 VSIZE=7 HSIZE=10
              NOCHARACTERS VPOS=45 HPOS=90;
PROC GCONTOUR;
      PLOT T*C=N/LEVELS=5 10 20 TO 160 BY 20
           LEVELS=22 21 20 10 9 8 4 3 2 1
           CLEVELS='GREEN' 'CYAN' 'BLUE' 'PINK' 'RED' 'YELLOW'
                   'CYAN' 'RED' 'GREEN' 'YELLOW'
           CTEXT=BLUE CAXIS=GREEN;
      FOOTNOTE .H=2.0 .F=TITALIC .C=YELLOW MAXWELL DISTRIBUTION .H=
              1.5 .F=DUPLEX .C=CYAN (NITROGEN);
```

プログラムF

```

DATA; SET DDD.MAXWELL;
GOPTIONS DEVICE=F6653 VSIZE=7 HSIZE=10
NOCHARACTERS VPOS=45 HPOS=90;
PROC GCONTOUR;
PLOT T*C=N/LEVELS=5 10 20 TO 160 BY 20 PATTERN JOIN
CTEXT=BLUE CAXIS=GREEN;
PATTERN1 C=YELLOW V=M4N90;
PATTERN2 C=CYAN V=M4N90;
PATTERN3 C=BLUE V=M4N90;
PATTERN4 C=PINK V=M4N90;
PATTERN5 C=GREEN V=M4N90;
PATTERN6 C=YELLOW V=M5N90;
PATTERN7 C=CYAN V=M5N90;
PATTERN8 C=BLUE V=M5N90;
PATTERN9 C=PINK V=M5N90;
PATTERN10 C=GREEN V=M5N90;
FOOTNOTE .H=2.0 .F=TITALIC .C=RED MAXWELL DISTRIBUTION .H=
1.5 .F=DUPLEX .C=PINK (NITROGEN);

```

プログラムG

```

DATA; SET DDD.MAXWELL;
GOPTIONS DEVICE=F6653 VSIZE=7 HSIZE=10
NOCHARACTERS VPOS=45 HPOS=90;
PROC G3D;
PLOT T*C=N/ROTATE=30 TILT=80 FANCY GRID LABSIZE=1.5
CAXIS=WHITE CTEXT=RED CTOP=GREEN CBOTTOM=BLUE;

```

プログラムH

プログラムFはプログラム17の変形である。CLEVELSは等高線の色指定である。プログラムEとは色の指定法が異なることに注意されたい。

プログラムGはプログラム18の変形である。PATTERNステートメントのCオプションは、各ハッチング・パターンの色を指定している。

プログラムHはプログラム22の変形である。CTOP, CBOTTOMオプションは鳥瞰図の表面と裏面の色指定である。


付録6 NLP出力の種々の実行法

SASの作画結果をNLPへ出力する場合の実行法および出力要求方法を下図に示す。

太線の経路が2.2で示した方法である。

その他の方法についての参考文献は次のとおりである。

B_SASJOB	...	文献1の第4章
SP_PLOT(GDP)	...	文献1の3.2
B_GDPCNV	...	文献8の2.3、文献10の2.3
GDPモード	...	文献8の第4章、文献10の第3章
B_PSPCNV	...	文献8の2.3、文献10の2.2
SORP	...	文献8の2.4、文献11の第4章
KOUT	...	文献8の2.5、文献12の6.3



付録 7 プログラムおよびデータの複写方法

この手引きに掲載されたすべてのプログラムは、次のデータセットに保存されている。

`'KYOIKU.EXP.SAS'`

メンバ名は、プログラム番号と対応しており、先頭の2文字をKWとし、その後プログラム番号を2桁の数字で付加する。例えば、プログラム7の保存されているメンバ名は、KW07となる。またプログラム7-2の場合は、KW07#2となる。付録に掲載されているものも同様に、例えば、プログラムEは、KWOEに保存されている。

また、作図に必要なデータは、次のデータセットに保存されている。

`'KYOIKU.EXP.DATA'`

この手引きで用いたデータは3種類であり、それらは、それぞれメンバ名KWDT1, KWDT2, KWDT3に保存されている。下表に、プログラムとデータの対応を示す。

データ	プログラム
KWDT1	1, 2, 3, 4, 5, 6, 7, 7-2, 8, 9, 9-2, 10, 15
KWDT2	11, 12, 13, 14, 14-2, 14-3, 14-4, D, E
KWDT3	7-3
不要	2-2, 16, 17, 24, 24-3, 25, A, B, C
SASデータセット	18, 19, 20, 21, 22, 23, F, G, H

なお、最下段のSASデータセットとは、プログラム17で作成した、各ユーザのSASデータセットを用いることを示している。DATAステップをプログラム16と同じにすれば、データは不要となる。

これらをユーザのデータセットに複写するには、次のコマンドを用いる。複写したいメンバ名を、例えばKWXXとし、複写先のユーザのデータセット名を、AAA.DATA(MMM)とすれば、

```
COPY 'KYOIKU.EXP.SAS(KWXX)' AAA.DATA(MMM)
```

ただし、複写したいものがデータの場合には、EXP.SASをEXP.DATAと変更しなければならない。

なお、PFD(文献14参照)を用いれば、複数のメンバを選択して複写することができる。

参考文献

SAS 関係の本センター発行の利用の手引き（利用の手引き特集 SAS 編にまとめられている）。

- 1) SAS（統計分析・図形処理）使用方法 車古正樹著（MNSV01）
- 2) SASによる統計分析（その1）—プログラムの入力・実行から出力まで— 田中勝人著（MNYV01）
- 3) SASによる統計分析（その2）—データ分析の実際— 田中勝人著（MNYV02）

SAS 関係のその他の文献

- 4) SAS/GRAPH USER'S GUIDE（日本語版） 1981 EDITION（英語版は内容が古いので注意されたい）
- 5) SAS USER'S GUIDE BASIC
- 6) SASによるデータ解析 基礎編（上） 中山和彦編
- 7) SASによるデータ解析 基礎編（下） 中山和彦編

直接 SAS とは関係しないその他の文献

- 8) 利用の手引き 計算機にグラフを書かせてみよう—GRACE IVを用いたプログラミングから検索、出力までの実際— 中野康英著（MNTV01）
- 9) 利用の手引き グラフィックディスプレイによる会話図形処理入門 — GSP 用法を中心に — 山崎光悦著（MNPV01）
- 10) 利用の手引き TSS実習書（全般用） TSSユーザのための便利なコマンド（その1） 車古正樹著（MNKV01）（利用の手引き特集 TSS編に含まれている）
- 11) 利用の手引き TSS端末によるバッチ処理—フルスクリーン機能と出力検索— 関崎正夫著（MNHV01）（利用の手引き特集 TSS編に含まれている）
- 12) 利用の手引き 日本語文章処理システム入門（その1）—日本語エディタと入出力方法— 車古正樹著（MNVV01）
- 13) SSL II使用手引書（科学用サブルーチンライブラリ）
- 14) 利用の手引き PFD入門—画面を利用した簡単な計算機使用法— 久米田 稔著（MNCW01）

追 記

本手引きの校正中に、次のことに気付いたので、書き加えておく。

3.14で紹介したSYMBOL文のI=SMnオプションは、本手引き中のようにXのある値に対して繰り返しのあるデータに関してのみ有効なオプションではない。通常の（例えば3.1～3.7のような）データ点に対しても有効である。その場合は、3.6や3.7のようにすべてのデータ点を通る線でもなく、また3.9のようにはっきり次数の決った線でもない、人間が通常行うような適当にばらつきを考慮した線を引いてくれる。